

The Agda Universal Algebra Library

Part 1: Foundation

Equality, extensionality, truncation, and dependent types for relations and algebras

William DeMeo   

Department of Algebra, Charles University in Prague

Abstract

The Agda Universal Algebra Library (UALib) is a library of types and programs (theorems and proofs) we developed to formalize the foundations of universal algebra in dependent type theory using the Agda programming language and proof assistant. The UALib includes a substantial collection of definitions, theorems, and proofs from general algebra and equational logic, including many examples that exhibit the power of inductive and dependent types for representing and reasoning about relations, algebraic structures, and equational theories. In this paper we discuss the logical foundations on which the library is built, and describe the types defined in the first 13 modules of the library. Special attention is given to aspects of the library that seem most interesting or challenging from a type theory or mathematical foundations perspective.

2012 ACM Subject Classification Theory of computation \rightarrow Logic and verification; Computing methodologies \rightarrow Representation of mathematical objects; Theory of computation \rightarrow Type theory

Keywords and phrases Agda, constructive mathematics, dependent types, equational logic, formalization of mathematics, model theory, type theory, universal algebra

Related Version hosted on arXiv

Part 2, Part 3: http://arxiv.org/a/demeo_w_1

Supplementary Material

Documentation: ualib.org

Software: <https://gitlab.com/ualib/ualib.gitlab.io.git>

Acknowledgements

The author thanks Cliff Bergman, Hyeyoung Shin, and Siva Somayyajula for supporting and contributing to this project, Adreas Abel for helpful corrections, and three anonymous referees for invaluable feedback on an early draft of this work. Thanks are also owed to [Martín Escardó](#) for creating the [Type Topology](#) library and teaching us about it at the [2019 Midlands Graduate School in Computing Science](#) [10]. Finally, the [AgdaUALib](#) would not exist in its current form without the [Agda](#) language, for which we thank the Agda Team (Andreas Abel, Guillaume Allais, Liang-Ting Chen, Jesper Cockx, Nils Anders Danielsson, Víctor López Juan, Ulf Norell, Andrés Sicard-Ramírez, Andrea Vezzosi, and Tesla Ice Zhang).



This work and the Agda Universal Algebra Library by [William DeMeo](#) is licensed under a [Creative Commons Attribution-ShareAlike 4.0 International License](#).



© 2021 [William DeMeo](#). Based on work at <https://gitlab.com/ualib/ualib.gitlab.io>.
Compiled with [xelatex](#) on 12 Apr 2021 at 22:55.

Contents

1	Introduction	2
1.1	Motivation	2
1.2	Attributions and Contributions	3
1.3	Prior art	3
1.4	Organization of the paper	3
1.5	Resources	4
2	Overture	5
2.1	Preliminaries: logical foundations, universes, dependent types	5
2.2	Equality: definitional equality and transport	9
2.3	Extensionality: types for postulating function extensionality	11
2.4	Inverses: epics, monics, embeddings, inverse images	13
2.5	Lifts: making peace with a noncumulative universe hierarchy	15
3	Relation Types	16
3.1	Discrete relations: predicates, axiom of extensionality, compatibility	16
3.2	Continuous relations: arbitrary-sorted relations of arbitrary arity* ¹	21
3.3	Quotients: equivalences, class representatives, quotient types	23
3.4	Truncation: continuous propositions, quotient extensionality	25
4	Algebra Types	29
4.1	Signatures: types for operations & signatures	30
4.2	Algebras: types for algebras, operation interpretation & compatibility	31
4.3	Products: types for products over arbitrary classes	33
4.4	Congruences: types for congruences & quotient algebras	34
5	Concluding Remarks	36
A	Dependency Graph	39

1 Introduction

To support formalization in type theory of research level mathematics in universal algebra and related fields, we present the Agda Universal Algebra Library ([AgdaUALib](#)), a software library containing formal statements and proofs of the core definitions and results of universal algebra. The [UALib](#) is written in [Agda](#) [17], a programming language and proof assistant based on Martin-Löf Type Theory (MLTT) that supports dependent and inductive types.

1.1 Motivation

The seminal idea for the [AgdaUALib](#) project was the observation that, on the one hand, a number of fundamental constructions in universal algebra can be defined recursively, and theorems about them proved by structural induction, while, on the other hand, inductive and dependent types make possible very precise formal representations of recursively defined objects, which often admit elegant constructive proofs of properties of such objects. An important feature of such proofs in type theory is that they are total functional programs and, as such, they are computable, composable, and machine-verifiable.

Finally, our own research experience has taught us that a proof assistant and programming language (like Agda), when equipped with specialized libraries and domain-specific tactics to automate the proof idioms of our field, can be an extremely powerful and effective asset. As such we believe that proof assistants and their supporting libraries will eventually become indispensable tools in the working mathematician’s toolkit.

1.2 Attributions and Contributions

The mathematical results described in this paper have well known *informal* proofs. Our main contribution is the formalization, mechanization, and verification of the statements and proofs of these results in dependent type theory using Agda.

Unless explicitly stated otherwise, the Agda source code described in this paper is due to the author, with the following caveat: the `UALib` depends on the `Type Topology` library of Martín Escardó [10]. For convenience, we refer to Escardó’s library as `TypeTopo` throughout the paper. For the sake of completeness and clarity, and to keep the paper mostly self-contained, we repeat some definitions from `TypeTopo`, but in each instance we cite the original source.²

1.3 Prior art

There have been a number of efforts to formalize parts of universal algebra in type theory prior to ours, most notably

- Capretta [4] (1999) formalized the basics of universal algebra in the Calculus of Inductive Constructions using the Coq proof assistant;
- Spitters and van der Weegen [19] (2011) formalized the basics of universal algebra and some classical algebraic structures, also in the Calculus of Inductive Constructions using the Coq proof assistant, promoting the use of type classes;
- Gunther, et al [11] (2018) developed what seems to be (prior to the `UALib`) the most extensive library of formal universal algebra to date; in particular, this work includes a formalization of some basic equational logic; also (unlike the `UALib`) it handles *multisorted* algebraic structures; (like the `UALib`) it is based on dependent type theory and the Agda proof assistant.

Some other projects aimed at formalizing mathematics generally, and algebra in particular, have developed into very extensive libraries that include definitions, theorems, and proofs about algebraic structures, such as groups, rings, modules, etc. However, the goals of these efforts seem to be the formalization of special classical algebraic structures, as opposed to the general theory of (universal) algebras. Moreover, the part of universal algebra and equational logic formalized in the `UALib` extends beyond the scope of prior efforts. In particular, the library now includes a proof of Birkhoff’s variety theorem. Most other proofs of this theorem that we know of are informal and nonconstructive.³

1.4 Organization of the paper

In this paper we limit ourselves to the presentation of the core foundational modules of the `UALib` so that we have space to discuss some of the more interesting type theoretic and found-

² In the `UALib`, such instances occur only inside hidden modules that are never actually used, followed immediately by a statement that imports the code in question from its original source.

³ After completing the formal proof in Agda, we learned about a constructive version of Birkhoff’s theorem proved by Carlström in [5]. The latter is presented in the informal style of standard mathematical writing, and as far as we know it was never formalized in type theory and type-checked with a proof assistant. Nonetheless, a comparison of Carlström’s proof and the `UALib` proof would be interesting.

ational issues that arose when developing the library and attempting to represent advanced mathematical notions in type theory and formalize them in Agda. This is the first in a series of three papers describing the [AgdaUALib](#). The second paper ([8]) covers *homomorphisms*, *terms*, and *subalgebras*. The third paper ([9]) covers *free algebras*, *equational classes* of algebras (i.e., *varieties*), and *Birkhoff’s HSP theorem*.

This present paper is organized into three parts as follows. The first part is §2 which introduces the basic concepts of type theory with special emphasis on the way such concepts are formalized in [Agda](#). Specifically, §2.1 introduces *Sigma types* and Agda’s hierarchy of *universes*. The important topics of *equality* and *function extensionality* are discussed in §2.2 and §2.3; §2.4 covers inverses and inverse images of functions. In §2.5 we describe a technical problem that one frequently encounters when working in a *noncumulative universe hierarchy* and offer some tools for resolving the type-checking errors that arise from this.

The second part is §3 which covers *relation types* and *quotient types*. Specifically, §3.1 defines types that represent *unary* and *binary relations* as well as *function kernels*. These “discrete relation types,” are all very standard. In §3.2 we introduce the (less standard) types that we use to represent *general* and *dependent relations*. We call these “continuous relations” because they can have arbitrary arity (general relations) and they can be defined over arbitrary families of types (dependent relations). In §3.3 we cover standard types for equivalence relations and quotients, and in §3.4 we discuss a family of concepts that are vital to the mechanization of mathematics using type theory; these are the closely related concepts of *truncation*, *sets*, *propositions*, and *proposition extensionality*.

The third part of the paper is §4 which covers the basic domain-specific types offered by the [UALib](#). It is here that we finally get to see some types representing algebraic structures. Specifically, we describe types for *operations* and *signatures* (§4.1), *general algebras* (§4.2), and *product algebras* (§4.3), including types for representing *products over arbitrary classes of algebraic structures*. Finally, we define types for congruence relations and quotient algebras in §4.4.

1.5 Resources

We conclude this introduction with some pointers to helpful reference materials. For the required background in Universal Algebra, we recommend the textbook by Clifford Bergman [1]. For the type theory background, we recommend the HoTT Book [18] and Escardó’s [Introduction to Univalent Foundations of Mathematics with Agda](#) [10].

The following are informed the development of the [UALib](#) and are highly recommended.

- [Introduction to Univalent Foundations of Mathematics with Agda](#), Escardó [10].
- [Dependent Types at Work](#), Bove and Dybjer [2].
- [Dependently Typed Programming in Agda](#), Norell and Chapman [16].
- [Formalization of Universal Algebra in Agda](#), Gunther, Gadea, Pagano [11].
- [Programming Languages Foundations in Agda](#), Philip Wadler [24].

More information about [AgdaUALib](#) can be obtained from the following official sources.

- [ualib.org](#) (the web site) documents every line of code in the library.
- [gitlab.com/ualib/ualib.gitlab.io](#) (the source code) [AgdaUALib](#) is open source.⁴
- [The Agda UALib, Part 2: homomorphisms, terms, and subalgebras](#) [8].
- [The Agda UALib, Part 3: free algebras, equational classes, and Birkhoff’s theorem](#) [9].

⁴ License: [Creative Commons Attribution-ShareAlike 4.0 International License](#).

The first item links to the official [UALib](#) html documentation which includes complete proofs of every theorem we mention here, and much more, including the Agda modules covered in the first and third installments of this series of papers on the [UALib](#).

Finally, readers will get much more out of reading the paper if they download the [AgdaUALib](#) from <https://gitlab.com/ualib/ualib.gitlab.io>, install the library, and try it out for themselves.

2 Overture

2.1 Preliminaries: logical foundations, universes, dependent types

This section presents the [Overture.Preliminaries](#) module of the [AgdaUALib](#), slightly abridged.⁵ Here we define or import the basic types of *Martin-Löf type theory* ([MLTT](#)). Although this is standard stuff, we take this opportunity to highlight aspects of the [UALib](#) syntax that may differ from that of “standard Agda.”

2.1.1 Logical foundations

The [AgdaUALib](#) is based on a type theory that is the same or very close to the one on which on which Martín Escardó’s [Type Topology](#) ([TypeTopo](#)) Agda library is based. We don’t discuss [MLTT](#) in great detail here because there are already nice and freely available resources covering the theory. (See, for example, the section [A spartan Martin-Löf type theory](#) of the lecture notes by Escardó [10], the [ncatlab entry on Martin-Löf dependent type theory](#), or the [HoTT Book](#) [18].)

The objects and assumptions that form the foundation of [MLTT](#) are few. There are the *primitive types* ($\mathbf{0}$, $\mathbf{1}$, and \mathbf{N} , denoting the empty type, one-element type, and natural numbers), the *type formers* ($+$, Π , Σ , Id , denoting *binary sum*, *product*, *sum*, and the *identity* type). Each of these type formers is defined by a *type forming rule* which specifies how that type is constructed. Lastly, we have an infinite collection of *type universes* (types of types) and *universe variables* to denote them. Following Escardó, we denote universes in the [UALib](#) by upper-case calligraphic letters from the second half of the English alphabet; to be precise, these are \mathcal{O} , \mathcal{Q} , \mathcal{R} , \dots , \mathcal{X} , \mathcal{Y} , \mathcal{Z} .⁶

That’s all. There are no further axioms or logical deduction (proof derivation) rules needed for the foundation of [MLTT](#) that we take as the starting point of the [AgdaUALib](#). The logical semantics come from the [propositions-as-types correspondence](#) [15]: propositions and predicates are represented by types and the inhabitants of these types are the proofs of the propositions and predicates. As such, proofs are constructed using the type forming rules. In other words, the type forming rules *are* the proof derivation rules.

To this foundation, we add certain *extensionality principles* when and where we need them. These will be developed as we progress. However, classical axioms such as the [Axiom of Choice](#) or the [Law of the Excluded Middle](#) are not needed and are not assumed anywhere in the library. In that sense, all theorems and proofs in the [UALib](#) are [constructive](#) (as defined, e.g., in [13]).

A few specific instances (e.g., the proof of the Noether isomorphism theorems and Birkhoff’s HSP theorem) require certain *truncation* assumptions. In such cases, the theory is not [predicative](#) (as defined, e.g., in [14]). These instances are always clearly identified.

⁵ For unabridged docs and source code see <https://ualib.gitlab.io/Overture.Preliminaries.html> and <https://gitlab.com/ualib/ualib.gitlab.io/-/blob/master/UALib/Overture/Preliminaries.lagda>.

⁶ We avoid using \mathcal{P} as a universe variable because it is used to denote the powerset type in [TypeTopo](#).

Specifying logical foundations in Agda

An Agda program typically begins by setting some options and by importing types from existing Agda libraries. Options are specified with the `OPTIONS` pragma and control the way Agda behaves, for example, by specifying the logical axioms and deduction rules we wish to assume when the program is type-checked to verify its correctness. Every Agda program in the `UALib` begins with the following line.

```
{-# OPTIONS -without-K -exact-split -safe #-}
```

 (1)

These options control certain foundational assumptions that Agda makes when type-checking the program to verify its correctness.

1. `-without-K` disables [Streicher’s K axiom](#), which makes Agda compatible with *proof-relevant* type theories; see the discussion of proof-relevance in § 3.4; see also [20, 7];
2. `-exact-split` makes Agda accept only definitions that are *judgmental* equalities; see [22];
3. `-safe` ensures that nothing is postulated outright—every non-`MLTT` axiom has to be an explicit assumption (e.g., an argument to a function or module); see [21, 22].

Throughout this paper we take assumptions 1–3 for granted without mentioning them explicitly.

2.1.2 Agda Modules

The `OPTIONS` pragma is usually followed by the start of a module. For example, the `Overture.Preliminaries` module begins with the following line.

```
module Overture.Preliminaries where
```

Sometimes we want to declare parameters that will be assumed throughout the module. For instance, when working with algebras, we often assume they come from a particular fixed signature, and this signature is something we could fix as a parameter at the start of a module. Thus, we might start an *anonymous submodule* of the main module with a line like⁷

```
module _ {S : Signature ⊆ V} where
```

Such a module is called *anonymous* because an underscore appears in place of a module name. Agda determines where a submodule ends by indentation. This can take some getting used to, but after a short time it will feel very natural. The main module of a file must have the same name as the file, without the `.agda` or `.lagda` file extension. The code inside the main module is not indented. Submodules are declared inside the main module and code inside these submodules must be indented to a fixed column. As long as the code is indented, Agda considers it part of the submodule. A submodule is exited as soon as a nonindented line of code appears.

2.1.3 Universes in Agda

For the very small amount of background we require about the notion of *type universe* (or *level*), we refer the reader to the brief [section on universe-levels](#) in the [Agda documentation](#).⁸

Throughout the `AgdaUALib` we use many of the nice tools that Martín Escardó has developed and made available in `TypeTopo` library of Agda code for the *Univalent Foundations* of math-

⁷ The `Signature` type will be defined in Section 4.1.

⁸ See <https://agda.readthedocs.io/en/v2.6.1.3/language/universe-levels.html>.

ematics.⁹ The first of these is the `Universes` module which we import as follows.

```
open import Universes public
```

Since we use the `public` directive, the `Universes` module will be available to all modules that import the present module (`Overture.Preliminaries`). This module declares symbols used to denote universes. As mentioned, we adopt Escardó’s convention of denoting universes by capital calligraphic letters, and most of the ones we use are already declared in `Universes`; those that are not are declared as follows.

```
variable  $\mathcal{U} \mathcal{V} \mathcal{X} : \text{Universe}$ 
```

The `Universes` module also provides alternative syntax for the primitive operations on universes that Agda supports. Specifically, the `·` operator maps a universe level \mathcal{U} to the type `Set \mathcal{U}` , and the latter has type `Set (Isuc \mathcal{U})`. The Agda level `lzero` is renamed \mathcal{U}_0 , so $\mathcal{U}_0 \cdot$ is an alias for `Set lzero`. Thus, $\mathcal{U} \cdot$ is simply an alias for `Set \mathcal{U}` , and we have `Set $\mathcal{U} : \text{Set (Isuc \mathcal{U})}$` . Finally, `Set (Isuc lzero)` is equivalent to `Set \mathcal{U}_0^+` , which we (and Escardó) denote by $\mathcal{U}_0^+ \cdot$.

To justify the introduction of this somewhat nonstandard notation for universe levels, Escardó points out that the Agda library uses `Level` for universes (so what we write as $\mathcal{U} \cdot$ is written `Set \mathcal{U}` in standard Agda), but in univalent mathematics the types in $\mathcal{U} \cdot$ need not be sets, so the standard Agda notation can be a bit confusing, especially to newcomers.

There will be many occasions calling for a type living in a universe at the level that is the least upper bound of two universe levels, say, \mathcal{U} and \mathcal{V} . The universe level $\mathcal{U} \sqcup \mathcal{V}$ denotes this least upper bound. Here \sqcup is an Agda primitive designed for precisely this purpose.

2.1.4 Dependent types

Sigma types (dependent pairs)

Given universes \mathcal{U} and \mathcal{V} , a type $A : \mathcal{U} \cdot$, and a type family $B : A \rightarrow \mathcal{V} \cdot$, the *Sigma type* (or *dependent pair type*, or *dependent product type*) is denoted by $\Sigma x : A, B x$ and generalizes the Cartesian product $A \times B$ by allowing the type $B x$ of the second argument of the ordered pair (x, y) to depend on the value x of the first. That is, an inhabitant of the type $\Sigma x : A, B x$ is a pair (x, y) such that $x : A$ and $y : B x$.

The dependent product type is defined in `TypeTopo` in a standard way. For pedagogical purposes we repeat the definition here.¹⁰

```
record  $\Sigma \{ \mathcal{U} \mathcal{V} \} \{ A : \mathcal{U} \cdot \} (B : A \rightarrow \mathcal{V} \cdot) : \mathcal{U} \sqcup \mathcal{V} \cdot \text{ where}$ 
  constructor _,_
  field
    pr1 : A
    pr2 : B pr1
```

Agda’s default syntax for this type is $\Sigma \lambda(x : A) \rightarrow B$, but we prefer the notation $\Sigma x : A, B$,

⁹ Escardó has written an outstanding set of notes called [Introduction to Univalent Foundations of Mathematics with Agda](#), which we highly recommend to anyone looking for more details than we provide here about `MLTT` and Univalent Foundations/HoTT in Agda. [10].

¹⁰ In the `UALib` we put such redundant definitions inside “hidden” modules so that they doesn’t conflict with the original definitions which we import and use. It may seem odd to define something in a hidden module only to import and use an alternative definition, but we do this in order to exhibit all of the types on which the `UALib` depends while ensuring that this cannot be misinterpreted as a claim to originality.

which is closer to the syntax in the preceding paragraph, and will be familiar to readers of the HoTT book [18], for example. Fortunately, `TypeTopo` makes the preferred notation available with the following type definition and `syntax` declaration (see [10, Σ types]).¹¹

```
-Σ : {U V : Universe} (A : U ·) (B : A → V ·) → U ⊔ V ·
-Σ A B = Σ B

syntax -Σ A (λ x → B) = Σ x : A , B
```

A special case of the Sigma type is the one in which the type B doesn't depend on A . This is the usual Cartesian product, defined in Agda as follows.

```
__×__ : U · → V · → U ⊔ V ·
A × B = Σ x : A , B
```

Pi types (dependent functions)

Given universes \mathcal{U} and \mathcal{V} , a type $A : \mathcal{U} \cdot$, and a type family $B : A \rightarrow \mathcal{V} \cdot$, the *Pi type* (or *dependent function type*) is denoted by $\Pi x : A , B x$ and generalizes the function type $A \rightarrow B$ by letting the type $B x$ of the codomain depend on the value x of the domain type. The dependent function type is defined in `TypeTopo` in a standard way. For the reader's benefit, however, we repeat the definition here.

```
Π : {A : U ·} (A : A → V ·) → U ⊔ V ·
Π {A} A = (x : A) → A x
```

To make the syntax for Π conform to the standard notation for Pi types, Escardó uses the same trick as the one used above for Sigma types.¹¹

```
-Π : (A : U ·) (B : A → V ·) → U ⊔ V ·
-Π A B = Π B

syntax -Π A (λ x → b) = Π x : A , b
```

Once we have studied the types (defined in `TypeTopo` and repeated here for convenience and illustration purposes), the original definitions are imported like so.

```
open import Sigma-Type public
open import MGS-MLTT using (pr1; pr2; __×__; -Σ; Π; -Π) public
```

Projection notation

The definition of Σ (and thus \times) includes the fields `pr1` and `pr2` representing the first and second projections out of the product. Sometimes we prefer to denote these projections by `|_` and `||_`, respectively. However, for emphasis or readability we alternate between these and the following standard notations: `pr1` and `fst` for the first projection, `pr2` and `snd` for the second. We define these alternative notations for projections as follows.

```
module _ {U : Universe} {A : U ·} {B : A → V ·} where

  |_ fst : Σ B → A
```

¹¹ **Attention!** The symbol `:` that appears in the special syntax defined here for the Σ type, and below for the Π type, is not the ordinary colon; rather, it is the symbol obtained by typing `\:4` in `agda2-mode`.


```

| x , y | = x
fst (x , y) = x

||_|| snd : (z :  $\Sigma$  B)  $\rightarrow$  B (pr1 z)
|| x , y || = y
snd (x , y) = y

```

Remarks.

- We place these definitions (of `|_`, `fst`, `||_||` and `snd`) inside an *anonymous module*, which is a module that begins with the `module` keyword followed by an underscore character (instead of a module name). The purpose is to move some of the postulated typing judgments—the “parameters” of the module (e.g., `U : Universe`)—out of the way so they don’t obfuscate the definitions inside the module. In library documentation, such as the present paper, we often omit such module directives. In contrast, the collection of html pages at ualib.org, which is the most current and comprehensive documentation of the UALib, omits nothing.
- As the four definitions above make clear, multiple inhabitants of a single type (e.g., `|_` and `fst`) may be declared on the same line.

2.2 Equality: definitional equality and transport

This section presents the `Overture.Equality` module of the `AgdaUALib`, slightly abridged.¹²

2.2.1 Definitional equality

Here we discuss the basic but important type of MLTT called *definitional equality*. This concept is most understood, at least heuristically, with the following slogan: “Definitional equality is the substitution-preserving equivalence relation generated by definitions.” We will make this precise below, but first let us quote from a primary source. Per Martin-Löf offers the following definition in [12, §1.11] (italics added):¹³

Definitional equality is defined to be the equivalence relation, that is, reflexive, symmetric and transitive relation, which is generated by the principles that a definiendum is always definitionally equal to its definiens and that definitional equality is preserved under substitution.

To be sure we understand what this means, let $:=$ denote the relation with respect to which x is related to y (denoted $x := y$) if and only if y is the *definition* of x . Then the definitional equality relation \equiv is the reflexive, symmetric, transitive, substitutive closure of $:=$. By *substitutive closure* we mean closure under the following *substitution rule*.

$$\frac{\{A : \mathcal{U} \cdot\} \{B : A \rightarrow \mathcal{W} \cdot\} \{x y : A\} \quad x \equiv y}{B x \equiv B y} \text{ (subst)}$$

The datatype used in the UALib to represent definitional equality is imported from the `Identity-Type` module of `TypeTopo`, but apart from superficial syntactic differences, it is equivalent to the standard *Paulin-Mohring style identity type* found in most other Agda libraries. We

¹²For unabridged docs and source code see <https://ualib.gitlab.io/Overture.Equality.html> and <https://gitlab.com/ualib/ualib.gitlab.io/-/blob/master/UALib/Overture/Equality.lagda>.

¹³The *definiendum* is the left-hand side of a defining equation, the *definiens* is the right-hand side. For readers who have never generated an equivalence relation: the *reflexive closure* of $R \subseteq A \times A$ is the union of R and all pairs of the form (a, a) ; the *symmetric closure* is the union of R and its inverse $\{(y, x) : (x, y) \in R\}$; we leave it to the reader to come up with the correct definition of transitive closure.

repeat the definition here for easy reference.

```
data ==_ {U} {A : U → U} : A → A → U → where refl : {x : A} → x == x
```

Whenever we need to complete a proof by simply asserting that x is *definitionally equal* to itself, we invoke `refl`. If we need to make explicit the implicit argument x , then we use `refl {x = x}`.

Assumed module contexts

Before proceeding, a word about a special convention we adopt in the sequel is in order. To reduce reader strain, we often omit easily inferred typing judgments which would normally appear in the list of parameters of a module or at the start of a type definition, though we sometimes make an announcement like the following (which applies to the present section):

Unless otherwise indicated, the prevailing context in this section is given by

```
module _ {U : Universe} {A : U → U} where
```

which means that all code in the current section is to be interpreted as occurring inside such an anonymous module, where the given typing judgments are taken for granted.

Identity is an equivalence

The relation `==` just defined is naturally an equivalence relation, and the formal proof of this fact is trivial. Indeed, we don't need to prove reflexivity, since that is the defining property of `==`, and the proofs of symmetry and transitivity are immediate.

```
==-sym : {x y : A} → x == y → y == x
==-sym refl = refl
```

```
==-trans : {x y z : A} → x == y → y == z → x == z
==-trans refl refl = refl
```

We prove that `==` obeys the substitution rule (`subst`) in the next section (see [ap §2.2.2](#)), but first we define some syntactic sugar that will make it easier to apply symmetry and transitivity of `==` in proofs.¹⁴

```
_-1 : {x y : A} → x == y → y == x
p-1 = ==-sym p
```

If we have a proof $p : x == y$, and we need a proof of $y == x$, then instead of `==-sym p` we can use the more intuitive p^{-1} . Similarly, the following syntactic sugar makes abundant appeals to transitivity easier to stomach.

```
_·_ : {x y z : A} → x == y → y == z → x == z
p · q = ==-trans p q
```

2.2.2 Transport

Alonzo Church characterized equality by declaring two things equal if and only if no property (predicate) can distinguish them (see [\[6\]](#)). In other terms, x and y are equal if and only if for

¹⁴**Unicode Hints** (`agda2-mode`): $\wedge^{-1} \rightsquigarrow ^{-1}$; $\mid \rightsquigarrow id$; $\rightsquigarrow \rightsquigarrow \cdot$. In general, for information about a character, place the cursor on the character and type `M-x describe-char` (or `M-x h d c`).

all P we have $P\ x \rightarrow P\ y$. One direction of this implication is sometimes called *substitution* or *transport* or *transport along an identity*. It asserts the following: if two objects are equal and one of them satisfies a given predicate, then so does the other. A type representing this notion is defined, along with the (polymorphic) identity function, in the `MGS-MLTT` module of `TypeTopo` as follows.¹⁵

```
id : {U : Universe} (A : U ·) → A → A
id A = λ x → x

transport : {A : U ·} (B : A → W ·) {x y : A} → x ≡ y → B x → B y
transport B (refl {x = x}) = id (B x)
```

A function is well-defined if and only if it maps equivalent elements to a single element and we often use this nature of functions in Agda proofs. It is equivalent to the substitution rule (subst) we defined in the last section. If we have a function $f : A \rightarrow B$, two elements $x\ y : A$ of the domain, and an identity proof $p : x \equiv y$, then we obtain a proof of $f\ x \equiv f\ y$ by simply applying the `ap` function like so, `ap f p : f x ≡ f y`. Escardó defines `ap` in `TypeTopo` as follows.

```
ap : {A : U ·} {B : V ·} (f : A → B) {a b : A} → a ≡ b → f a ≡ f b
ap f {a} p = transport (λ - → f a ≡ f -) p (refl {x = f a})
```

This establishes that our definitional equality satisfies the substitution rule (subst).

Here’s a useful variation of `ap` that we borrow from the `Relation/Binary/Core.agda` module of the `Agda Standard Library` (transcribed into `TypeTopo/UALib`).

```
cong-app : {A : U ·} {B : A → W ·} {f g : B} → f ≡ g → (a : A) → f a ≡ g a
cong-app refl _ = refl
```

2.3 Extensionality: types for postulating function extensionality

This section presents the `Overture.Extensionality` module of the `AgdaUALib`, slightly abridged.¹⁶

2.3.1 Background and motivation

This brief introduction to *function extensionality* is intended for novices. Those already familiar with the concept might wish to skip to the next subsection.

What does it mean to say that two functions $f\ g : X \rightarrow Y$ are equal? Suppose f and g are defined on $X = \mathbb{Z}$ (the integers) as follows: $f\ x := x + 2$ and $g\ x := ((2 * x) - 8) / 2 + 6$. Should we call f and g equal? Are they the “same” function? What does that even mean?

It’s hard to resist the urge to reduce g to $x + 2$ and proclaim that f and g are equal. Indeed, this is often an acceptable answer and the discussion normally ends there. In the science of computing, however, more attention is paid to equality, and with good reason.

We can probably all agree that the functions f and g above, while not syntactically equal, do produce the same output when given the same input so it seems fine to think of the functions as the same, for all intents and purposes. But we should ask ourselves at what point do we

¹⁵ Including every line of code of the `AgdaUALib` in this paper would result in an unbearable reading experience. We include all significant sections of code from the first 13 modules, but we omit lines indicating that redundant definitions of functions (e.g., `transport` and `ap`) occur inside named “hidden” modules. We also omit lines importing the original definitions of such duplicate definitions from `TypeTopo` library.

¹⁶ For unabridged docs and source code see <https://ualib.gitlab.io/Overture.Extensionality.html> and <https://gitlab.com/ualib/ualib.gitlab.io/-/blob/master/UALib/Overture/Extensionality.lagda>.

notice or care about the difference in the way functions are defined? What if we had started out this discussion with two functions f and g both of which take a list as argument and produce as output a correctly sorted version of the input list? Suppose f is defined using the [merge sort](#) algorithm, while g uses [quick sort](#). Probably few of us would call f and g the “same” in this case.

In the examples above, it is common to say that the two functions are *extensionally equal*, since they produce the same *external* output when given the same input, but they are not *intensionally equal*, since their *internal* definitions differ. In the next subsection we describe types that manifest this idea of *extensional equality of functions*, or *function extensionality*.¹⁷

2.3.2 Definition of function extensionality

As explained above, a natural notion of function equality is defined as follows: f and g are said to be *pointwise equal* provided $\forall x \rightarrow fx \equiv gx$. Here is how this is expressed in type theory (e.g., in [TypeTopo](#)).

$$\begin{aligned} _ \sim _ &: \{A : \mathcal{U} \cdot\} \{B : A \rightarrow \mathcal{W} \cdot\} \rightarrow \Pi B \rightarrow \Pi B \rightarrow \mathcal{U} \sqcup \mathcal{W} \cdot \\ f \sim g &= \forall x \rightarrow fx \equiv gx \end{aligned}$$

Function extensionality is the principle that pointwise equal functions are *definitionally equal*; that is, $\forall f g (f \sim g \rightarrow f \equiv g)$. In type theory this principle is represented by the types [funext](#) (for nondependent functions) and [dfunext](#) (for dependent functions) ([18, §2.9]). For example, the latter is defined as follows.

$$\begin{aligned} \text{dfunext} &: \forall \mathcal{U} \mathcal{W} \rightarrow (\mathcal{U} \sqcup \mathcal{W})^+ \cdot \\ \text{dfunext } \mathcal{U} \mathcal{W} &= \{A : \mathcal{U} \cdot\} \{B : A \rightarrow \mathcal{W} \cdot\} \{f g : \forall (x : A) \rightarrow B x\} \rightarrow f \sim g \rightarrow f \equiv g \end{aligned}$$

In informal settings, this so-called *point-wise equality of functions* is typically what one means when one asserts that two functions are “equal.”¹⁸ However, it is important to keep in mind the following fact: *function extensionality is known to be neither provable nor disprovable in Martin-Löf type theory. It is an independent statement.* [10]

Previous versions of the [UALib](#) made heavy use of a *global function extensionality principle* which asserts that function extensionality holds at all universe levels. However, we decided to remove all instances of global function extensionality from the latest version of the [UALib](#) and limit ourselves to local applications of the principle. This has the advantage of making transparent precisely how and where the library depends on function extensionality. The price we pay for this precision is a library that is littered with extensionality postulates.¹⁹

The next type defines a converse of function extensionality for dependent function types (cf. [cong-app](#) in [Overture.Equality](#) and [18, (2.9.2)]).

$$\begin{aligned} \text{happly} &: \{A : \mathcal{U} \cdot\} \{B : A \rightarrow \mathcal{W} \cdot\} \{f g : \Pi B\} \rightarrow f \equiv g \rightarrow f \sim g \\ \text{happly } _ _ \text{ refl } &= \text{ refl } \end{aligned}$$

Though it may seem obvious to some readers, we wish to emphasize the important conceptual

¹⁷ Most of these types are already defined in [TypeTopo](#), so the [UALib](#) imports the definitions from there; as usual, we redefine some of these types here for the purpose of explication.

¹⁸ In fact, if one assumes the *univalence axiom* of Homotopy Type Theory [18], then point-wise equality of functions is equivalent to definitional equality of functions. See the section “[Function extensionality from univalence](#)” of [10].

¹⁹ Eventually we will probably remove these postulates in favor of an alternative approach to extensionality; e.g., *univalence* and/or Cubical Agda.

distinction between two different forms of type definitions by comparing the definitions of `dfunext` and `happly`. In the definition of `dfunext`, the codomain is a parameterized type, namely, $\mathcal{U}^+ \sqcup \mathcal{V}^+ \cdot$, and the right-hand side of the defining equation of `dfunext` is an assertion (which may or may not hold). In the definition of `happly`, the codomain is an assertion, namely, $f \sim g$, and the right-hand side of the defining equation is a proof of this assertion. As such, `happly` is a *proof object*; it *proves* (or *inhabits the type that represents*) the proposition asserting that definitionally equivalent functions are pointwise equal. In contrast, `dfunext` is a type, and we may or may not wish to postulate an inhabitant of this type. That is, we could postulate that function extensionality holds by assuming we have a witness, say, $fe : \text{dfunext } \mathcal{U} \ \mathcal{V}$, but as noted above the existence of such a witness cannot be proved in MLTT.

Finally, a useful alternative for expressing dependent function extensionality, which is essentially equivalent to `dfunext`, is to assert that `happly` is actually an *equivalence* in a sense that we now describe. This requires a few definitions from the `MGS-Equivalences` module of `TypeTopo`. First, a type is a *singleton* if it has exactly one inhabitant and a *subsingleton* if it has at most one inhabitant.

```
is-center : (A :  $\mathcal{U} \cdot$ ) → A →  $\mathcal{U} \cdot$ 
is-center A c = (x : A) → c ≡ x

is-singleton :  $\mathcal{U} \cdot$  →  $\mathcal{U} \cdot$ 
is-singleton A =  $\Sigma$  c : A , is-center A c

is-subsingleton :  $\mathcal{U} \cdot$  →  $\mathcal{U} \cdot$ 
is-subsingleton A = (x y : A) → x ≡ y
```

Next, we consider the type `is-equiv` which is used to assert that a function is an equivalence in the sense that we now describe. This requires the concept of a *fiber* of a function, which can be represented as a Sigma type whose inhabitants denote inverse images of points in the codomain of the given function.

```
fiber : {A :  $\mathcal{U} \cdot$ } {B :  $\mathcal{W} \cdot$ } (f : A → B) → B →  $\mathcal{U} \sqcup \mathcal{W} \cdot$ 
fiber {A} f y =  $\Sigma$  x : A , f x ≡ y
```

A function is called an *equivalence* if all of its fibers are singletons.

```
is-equiv : {A :  $\mathcal{U} \cdot$ } {B :  $\mathcal{W} \cdot$ } → (A → B) →  $\mathcal{U} \sqcup \mathcal{W} \cdot$ 
is-equiv f =  $\forall$  y → is-singleton (fiber f y)
```

Finally we are ready to fulfill the promise of a type that provides an alternative means of postulating function extensionality.

```
hfunext :  $\forall$   $\mathcal{U} \ \mathcal{W} \rightarrow (\mathcal{U} \sqcup \mathcal{W})^+ \cdot$ 
hfunext  $\mathcal{U} \ \mathcal{W} = \{A : \mathcal{U} \cdot\} \{B : A \rightarrow \mathcal{W} \cdot\} (f g : \Pi B) \rightarrow \text{is-equiv } (\text{happly } f g)$ 
```

2.4 Inverses: epics, monics, embeddings, inverse images

This section presents the `Overture.Inverses` module of the `AgdaUALib`, slightly abridged.²⁰ Assume the following typing judgments: $\{\mathcal{U} \ \mathcal{W} : \text{Universe}\} \{A : \mathcal{U} \cdot\} \{B : \mathcal{W} \cdot\}$.

²⁰For unabridged docs and source code see <https://ualib.gitlab.io/Overture.Inverses.html> and <https://gitlab.com/ualib/ualib.gitlab.io/-/blob/master/UALib/Overture/Inverses.lagda>.

We begin by defining an inductive type that represents the *inverse image* of a function.

```
data Image_⊃_ (f : A → B) : B →  $\mathcal{U} \sqcup \mathcal{W}$  · where
  im : (x : A) → Image f ⊃ f x
  eq : (b : B) → (a : A) → b ≡ f a → Image f ⊃ b
```

Next we verify that the type behaves as we expect.

```
ImageIsImage : (f : A → B)(b : B)(a : A) → b ≡ f a → Image f ⊃ b
ImageIsImage f b a b≡fa = eq b a b≡fa
```

An inhabitant of $\text{Image } f \ni b$ is a pair (a, p) , where $a : A$, and p is a proof that f maps a to b ; that is, $p : b \equiv f a$. Since the proof that b belongs to the image of f is always accompanied by a witness $a : A$, we can actually *compute* a pseudoinverse of f . This will be a function that takes an arbitrary $b : B$ and a (*witness, proof*)-pair, $(a, p) : \text{Image } f \ni b$, and returns a .

```
Inv : (f : A → B){b : B} → Image f ⊃ b → A
Inv f {.(f a)} (im a) = a
Inv f (eq _ a _) = a
```

We can prove that $\text{Inv } f$ is the *right-inverse* of f , as follows.

```
InvIsInv : (f : A → B){b : B}(q : Image f ⊃ b) → f(Inv f q) ≡ b
InvIsInv f {.(f a)} (im a) = refl
InvIsInv f (eq _ _ p) = p-1
```

2.4.1 Epics (surjective functions)

We represent an *epic* (or *surjective*) function from A to B as an inhabitant of the following type.

```
Epic : (A → B) →  $\mathcal{U} \sqcup \mathcal{W}$  ·
Epic f =  $\forall y \rightarrow \text{Image } f \ni y$ 
```

With the next definition, we can represent the *right-inverse* of an epic function.

```
EpicInv : (f : A → B) → Epic f → B → A
EpicInv f fE b = Inv f (fE b)
```

The right-inverse of f is obtained by applying EpicInv to f along with a proof of $\text{Epic } f$. To see that this does indeed give the right-inverse we prove the EpicInvIsRightInv lemma below. This requires function composition, denoted \circ and defined in `TypeTopo` library.

```
_∘_ : {C : B →  $\mathcal{W}$  ·} →  $\Pi$  C → (f : A → B) → (x : A) → C (f x)
g ∘ f =  $\lambda x \rightarrow g (f x)$ 
```

Note that the next proof requires function extensionality, which we postulate a module declaration like this: `module _ { $\mathcal{U} \mathcal{W}$: Universe}{ f_e : funext $\mathcal{W} \mathcal{W}$ }{ $A : \mathcal{U} \cdot$ }{ $B : \mathcal{W} \cdot$ } where`

```
EpicInvIsRightInv : (f : A → B)(fE : Epic f) → f ∘ (EpicInv f fE) ≡ id B
EpicInvIsRightInv f fE = fE (λ x → InvIsInv f (fE x))
```

2.4.2 Monics (injective functions)

We say that a function $g : A \rightarrow B$ is *monic* (or *injective*) if it does not map distinct elements to a common point. The following types manifest this property and prove that monic functions

have left-inverses.

```

Monic : (g : A → B) →  $\mathcal{U} \sqcup \mathcal{W}$  ·
Monic g =  $\forall a_1 a_2 \rightarrow g a_1 \equiv g a_2 \rightarrow a_1 \equiv a_2$ 

MonicInv : (f : A → B) → Monic f → (b : B) → Image f  $\ni b \rightarrow A$ 
MonicInv f _ =  $\lambda b \text{ imfb} \rightarrow \text{Inv } f \text{ imfb}$ 

MonicInvsLeftInv : {f : A → B} {fM : Monic f} {x : A} → (MonicInv f fM)(f x)(im x)  $\equiv x$ 
MonicInvsLeftInv = refl

```

2.4.3 Embeddings

The `is-embedding` type is defined in `TypeTopo` in the following way.

```

is-embedding : (A → B) →  $\mathcal{U} \sqcup \mathcal{W}$  ·
is-embedding f =  $\forall b \rightarrow \text{is-subsingleton (fiber } f \text{ } b)$ 

```

Thus, `is-embedding` f asserts that f is a function all of whose fibers are subsingletons. Observe that an embedding is not simply an injective map. However, if we assume that the codomain B has *unique identity proofs* (UIP), then we can prove that a monic function into B is an embedding. We discuss the UIP principle in §3.4 where we present the `Relations.Truncation` module.

Finding a proof that a function is an embedding isn't always easy, but one path that is often straightforward is to first prove that the function is invertible and then invoke the following theorem.

```

invertibles-are-embeddings : (f : A → B) → invertible f → is-embedding f
invertibles-are-embeddings f fi = equivs-are-embeddings f (invertibles-are-equivs f fi)

```

Finally, embeddings are monic; from a proof $p : \text{is-embedding } f$ that f is an embedding we can construct a proof of `Monic` f . We confirm this as follows.

```

embedding-is-monic : (f : A → B) → is-embedding f → Monic f
embedding-is-monic f femb x y fxfy = ap pr1 ((femb (f x)) fx fy) where
  fx fy : fiber f (f x)
  fx = x , refl
  fy = y , (fxfy  $^{-1}$ )

```

2.5 Lifts: making peace with a noncumulative universe hierarchy

This section presents the `Overture.Lifts` module of the `AgdaUALib`, slightly abridged.²¹

2.5.1 Agda's universe hierarchy

The hierarchy of universes in Agda is structured as follows: $\mathcal{U} \cdot : \mathcal{U}^{+ \cdot}$, $\mathcal{U}^{+ \cdot} : \mathcal{U}^{++ \cdot}$, etc.²² This means that the universe $\mathcal{U} \cdot$ has type $\mathcal{U}^{+ \cdot}$, and $\mathcal{U}^{+ \cdot}$ has type $\mathcal{U}^{++ \cdot}$, and so on. It is important to note, however, this does *not* imply that $\mathcal{U} \cdot : \mathcal{U}^{++ \cdot}$. In other words, Agda's

²¹For unabridged docs and source code see <https://ualib.gitlab.io/Overture.Lifts.html> and <https://gitlab.com/ualib/ualib.gitlab.io/-/blob/master/UALib/Overture/Lifts.lagda>.

²²Recall, from the `Overture.Preliminaries` module (§2.1.3), the special notation we use to denote Agda's *levels* and *universes*.

universe hierarchy is *noncumulative*. This can be advantageous as it becomes possible to treat universe levels more generally and precisely. On the other hand, a noncumulative hierarchy can sometimes make it seem unduly difficult to convince Agda that a program or proof is correct. To help us overcome this technical issue, there are some general universe lifting and lowering functions, which we describe in the next subsection. In Section 4.2.4 we will define a couple of domain-specific analogs of these tools. Later, in the modules presented in [8, 9], we prove various properties that make these effective mechanisms for resolving universe level problems when working with algebra types.

2.5.2 Lifting and lowering

Let us be more concrete about what is at issue here by considering a typical example. Agda frequently encounters errors during the type-checking process and responds by printing an error message. Often the message has the following form.

```
Birkhoff.lagda:498,20-23
   $\mathcal{U} \text{ := } \mathcal{O} \sqcup \mathcal{V} \sqcup (\mathcal{U}^+)$  when checking that... has type...
```

This error message means that Agda encountered the universe \mathcal{U} on line 498 (columns 20–23) of the file `Birkhoff.lagda`, but was expecting to find the universe $\mathcal{O} \sqcup \mathcal{V} \sqcup \mathcal{U}^+$ instead.

The general `Lift` record type that we now describe makes these situations easier to deal with. It takes a type inhabiting some universe and embeds it into a higher universe and, apart from syntax and notation, it is equivalent to the `Lift` type one finds in the `Level` module of the `Agda Standard Library`.

```
record Lift { $\mathcal{W}$   $\mathcal{U}$  : Universe} (A :  $\mathcal{U}$   $\bullet$ ) :  $\mathcal{U} \sqcup \mathcal{W}$   $\bullet$  where
  constructor lift
  field lower : A
open Lift
```

The point of having a ramified hierarchy of universes is to avoid Russell’s paradox, and this would be subverted if we were to lower the universe of a type that wasn’t previously lifted. However, we can prove that if an application of `lower` is immediately followed by an application of `lift`, then the result is the identity transformation. Similarly, `lift` followed by `lower` is the identity.

```
lift~lower : { $\mathcal{W}$   $\mathcal{U}$  : Universe} {A :  $\mathcal{U}$   $\bullet$ } → lift ∘ lower ≡ id (Lift{ $\mathcal{W}$ } A)
lift~lower = refl

lower~lift : { $\mathcal{W}$   $\mathcal{U}$  : Universe} {A :  $\mathcal{U}$   $\bullet$ } → lower{ $\mathcal{W}$ } { $\mathcal{U}$ } ∘ lift ≡ id A
lower~lift = refl
```

The proofs are trivial. Nonetheless, we’ll come across some holes these types can fill.

3 Relation Types

3.1 Discrete relations: predicates, axiom of extensionality, compatibility

This section presents the `Relations.Discrete` module of the `AgdaUALib`, slightly abridged.²³ Here we present the submodules of the `AgdaUALib`’s `Relations` module. In §3.1 we define types that

²³For unabridged docs and source code see <https://ualib.gitlab.io/Relations.Discrete.html> and <https://gitlab.com/ualib/ualib.gitlab.io/-/blob/master/UALib/Relations/Discrete.lagda>.

represent *unary* and *binary relations*, which we refer to as “discrete relations” to contrast them with the (“continuous”) *general* and *dependent relations* that we introduce in §3.2. We call the latter “continuous relations” because they can have arbitrary arity (general relations) and they can be defined over arbitrary families of types (dependent relations).

3.1.1 Unary relations

In set theory, given two sets A and P , we say that P is a *subset* of A , and we write $P \subseteq A$, just in case $\forall x (x \in P \rightarrow x \in A)$. We need a mechanism for representing this notion in Agda. A typical approach is to use a *predicate* type, denoted by `Pred`.

Given two universes \mathcal{U} \mathcal{W} and a type $A : \mathcal{U}$, the type `Pred A \mathcal{W}` represents *properties* that inhabitants of A may or may not satisfy. We write $P : \text{Pred } A \mathcal{U}$ to represent the semantic concept of the collection of inhabitants of A that satisfy (or belong to) P . Here is the definition.²⁴

```
Pred :  $\mathcal{U} \rightarrow (\mathcal{W} : \text{Universe}) \rightarrow \mathcal{U} \sqcup \mathcal{W}^+.$ 
Pred A  $\mathcal{W} = A \rightarrow \mathcal{W}.$ 
```

Later we consider predicates over the class of algebras in a given signature. In the `Algebras` module we will define the type `Algebra \mathcal{U} S` of S -algebras with domain type \mathcal{U} , and the type `Pred (Algebra \mathcal{U} S) \mathcal{W}` will represent classes of S -algebras with certain properties.

3.1.2 Membership and inclusion relations

Like the `Agda Standard Library`, the `UALib` includes types that represent the *element inclusion* and *subset inclusion* relations from set theory. For example, given a predicate `P`, we may represent that “ x belongs to `P`” or that “ x has property `P`,” by writing either $x \in P$ or `P` x . The definition of \in is standard. Nonetheless, here it is.²⁴

```
_∈_ : A → Pred A  $\mathcal{W} \rightarrow \mathcal{W}.$ 
 $x \in P = P\ x$ 
```

The *subset* relation is denoted, as usual, with the \subseteq symbol and is defined as follows.²⁴

```
_⊆_ : Pred A  $\mathcal{W} \rightarrow \text{Pred } A \mathcal{X} \rightarrow \mathcal{U} \sqcup \mathcal{W} \sqcup \mathcal{X}.$ 
 $P \subseteq Q = \forall \{x\} \rightarrow x \in P \rightarrow x \in Q$ 
```

3.1.3 The axiom of extensionality

In type theory everything is represented as a type and, as we have just seen, this includes subsets. Equality of types is a nontrivial matter, and thus so is equality of subsets when represented as unary predicates. Fortunately, it is straightforward to write down a type that represents what it typically means in informal mathematics to say that two subsets are (extensionally) equal—namely, they contain the same elements. In the `UALib` we denote this type by $\dot{=}$ and define it as follows.²⁵

```
_⊆̇_ : Pred A  $\mathcal{W} \rightarrow \text{Pred } A \mathcal{X} \rightarrow \mathcal{U} \sqcup \mathcal{W} \sqcup \mathcal{X}.$ 
 $P \dot{=} Q = (P \subseteq Q) \times (Q \subseteq P)$ 
```

A proof of $P \dot{=} Q$ is a pair (p, q) where $p : P \subseteq Q$ and $q : Q \subseteq P$ are proofs of the first and

²⁴cf. `Relation/Unary.agda` in the `Agda Standard Library`.

²⁵**Unicode Hints.** In `agda2-mode`, $\backslash. = \rightsquigarrow \dot{=}$, $\backslash u+ \rightsquigarrow \mathbb{U}$, $\backslash b0 \rightsquigarrow \mathbf{0}$, $\backslash B0 \rightsquigarrow \mathbf{0}$, $\backslash b1 \rightsquigarrow \mathbf{1}$, $\backslash B1 \rightsquigarrow \mathbf{1}$.

second inclusions, respectively. If P and Q are definitionally equal (i.e., $P \equiv Q$), then both $P \subseteq Q$ and $Q \subseteq P$ hold, so $P \doteq Q$ also holds, as we now confirm.

```
Pred-≡ : {P Q : Pred A ℳ} → P ≡ Q → P ≐ Q
Pred-≡ refl = (λ z → z) , (λ z → z)
```

The converse is not provable in MLTT. However, we can postulate that it holds as an axiom if we wish. This is called the *axiom of extensionality* and a type that represents it is the following.

```
ext-axiom : ℳ → (ℳ : Universe) → ℳ ⊔ ℳ → ℳ
ext-axiom A ℳ = ∀ (P Q : Pred A ℳ) → P ≐ Q → P ≡ Q
```

Note that `ext-axiom` does not itself postulate the axiom of extensionality; it merely says what it is. If we want to postulate the axiom, we must assume we have an inhabitant (or “witness”) of the type. We could do this in Agda in a number of ways, but probably the easiest is to simply add the witness as a parameter to a module, like so.²⁶

```
module ext-axiom-postulated {ℳ ℳ' : Universe} {A : ℳ → ℳ'} {ea : ext-axiom A ℳ} where
```

Other notions of extensionality come up often in the UALib; see, for example, §2.3 and §3.4.

Predicates toolbox

Here is a small collection of tools that will come in handy later. The first is an inductive type that represents *disjoint union*.²⁵

```
data _⊔_ (A : ℳ → ℳ') (B : ℳ' → ℳ') : ℳ ⊔ ℳ' → ℳ' where
  inj₁ : (x : A) → A ⊔ B
  inj₂ : (y : B) → A ⊔ B
```

And this can be used to define a type representing *union*, as follows.

```
_∪_ : Pred A ℳ → Pred A ℳ' → Pred A (ℳ ⊔ ℳ')
P ∪ Q = λ x → x ∈ P ⊔ x ∈ Q
```

Next we define convenient notation for asserting that the image of a function (the first argument) is contained in a predicate (the second argument).

```
Im ⊆ : (A → B) → Pred B ℳ → ℳ ⊔ ℳ'
Im f ⊆ S = ∀ x → f x ∈ S
```

The *empty set* is naturally represented by the *empty type*, \emptyset , and the latter is defined in the `Empty-Type` module of `TypeTopo`.^{25,27}

```
∅ : Pred A ℳ₀
∅ _ = ∅
```

Before closing our little predicates toolbox, let’s add to it a type that provides a natural way to encode *singletons*.

²⁶ Agda also has a `postulate` mechanism that we could use, but this would require omitting the `-safe` pragma from the `OPTIONS` directive at the start of the module.

²⁷ The empty type is an inductive type with no constructors; that is, `data ∅ {ℳ} : ℳ → ℳ' where - (empty body)`.

```

{ _ } : A → Pred A _
{ x } = x ≡ _

```

3.1.4 Binary Relations

In set theory, a binary relation on a set A is simply a subset of the Cartesian product $A \times A$. As such, we could model such a relation as a (unary) predicate over the product type $A \times A$, or as an inhabitant of the function type $A \rightarrow A \rightarrow \mathcal{W}$ (for some universe \mathcal{W}). Note, however, this is not the same as a unary predicate over the function type $A \rightarrow A$ since the latter has type $(A \rightarrow A) \rightarrow \mathcal{W}$, while a binary relation should have type $A \rightarrow (A \rightarrow \mathcal{W})$.

A generalization of the notion of binary relation is a *relation from A to B* , which we define first and treat binary relations on a single A as a special case.

```

REL :  $\mathcal{U} \rightarrow \mathcal{W} \rightarrow (\mathcal{X} : \text{Universe}) \rightarrow \mathcal{U} \sqcup \mathcal{W} \sqcup \mathcal{X}^+ .
REL\ A\ B\ \mathcal{X} = A \rightarrow B \rightarrow \mathcal{X} .

Rel :  $\mathcal{U} \rightarrow (\mathcal{X} : \text{Universe}) \rightarrow \mathcal{U} \sqcup \mathcal{X}^+ .
Rel\ A\ \mathcal{X} = REL\ A\ A\ \mathcal{X}$$ 
```

The kernel of a function

The *kernel* of a function $f : A \rightarrow B$ is defined informally by $\{(x, y) \in A \times A : f\ x = f\ y\}$. This can be represented in type theory in a number of ways, each of which may be useful in a particular context. For example, we could define the kernel to be an inhabitant of a (binary) relation type, a (unary) predicate type, a (curried) Sigma type, or an (uncurried) Sigma type. The alternatives are defined in the UALib as follows.

```

ker : (A → B) → Rel A  $\mathcal{W}$ 
ker\ g\ x\ y = g\ x ≡ g\ y

kernel : (A → B) → Pred (A × A)  $\mathcal{W}$ 
kernel\ g\ (x, y) = g\ x ≡ g\ y

ker-sigma : (A → B) →  $\mathcal{U} \sqcup \mathcal{W} .
ker-sigma\ g = \Sigma\ x : A , \Sigma\ y : A , g\ x ≡ g\ y

ker-sigma' : (A → B) →  $\mathcal{U} \sqcup \mathcal{W} .
ker-sigma'\ g = \Sigma\ (x, y) : (A \times A) , g\ x ≡ g\ y$$ 
```

Similarly, the *identity relation* (which is equivalent to the kernel of an injective function) can be represented by a number of different types. Here we only show the representation that we use later to construct the zero congruence. The notation we use here is close to that of conventional algebra notation, where 0_A is used to denote the identity relation $\{(x, y) \in A \times A : x = y\}$.²⁵

```

0 : Rel A  $\mathcal{U}$ 
0\ x\ y = x ≡ y

```

Finally, the *total relation* over A , which in set theory is the full Cartesian product $A \times A$, can be represented using the one-element type from the Unit-Type module of TypeTopo as follows.^{25,28}

²⁸The one-element type is an inductive type with a single constructor, denoted \star , as follows: `data 1 { \mathcal{U} } : $\mathcal{U} . where \star : 1$.`

$\mathbf{1} : \text{Rel } A \mathcal{U}_0$
 $\mathbf{1} \ a \ b = \mathbf{1}$

The implication relation²⁹

The following types represent *implication* for binary relations.

$_\text{on}_ : (B \rightarrow B \rightarrow C) \rightarrow (A \rightarrow B) \rightarrow (A \rightarrow A \rightarrow C)$
 $R \text{ on } g = \lambda x y \rightarrow R (g x) (g y)$

$_\Rightarrow_\ : \text{REL } A B \mathcal{X} \rightarrow \text{REL } A B \mathcal{Y} \rightarrow \mathcal{U} \sqcup \mathcal{W} \sqcup \mathcal{X} \sqcup \mathcal{Y} \ .$
 $P \Rightarrow Q = \forall \{i j\} \rightarrow P \ i \ j \rightarrow Q \ i \ j$

These combine to give a nice, general implication operation.

$_\Rightarrow[_] \Rightarrow_\ : \text{Rel } A \mathcal{X} \rightarrow (A \rightarrow B) \rightarrow \text{Rel } B \mathcal{Y} \rightarrow \mathcal{U} \sqcup \mathcal{X} \sqcup \mathcal{Y} \ .$
 $P \Rightarrow[\ g] \Rightarrow Q = P \Rightarrow (Q \text{ on } g)$

Operation type and compatibility

Notation. For consistency and readability, throughout the `UALib` we reserve two universe variables for special purposes. The first of these is \mathcal{O} which shall be reserved for types that represent *operation symbols* (see `Algebras.Signatures`). The second is \mathcal{V} which we reserve for types representing *arities* of relations or operations.

Below we will define types that are useful for asserting and proving facts about *compatibility* of operations and relations, but first we need a general type with which to represent operations. Here is the definition, which we justify below.

—The type of operations

$\text{Op} : \mathcal{V} \ . \rightarrow \mathcal{U} \ . \rightarrow \mathcal{U} \sqcup \mathcal{V} \ .$
 $\text{Op } I A = (I \rightarrow A) \rightarrow A$

The definition of `Op` codifies the arity of an operation as an arbitrary type $I : \mathcal{V} \ .$, which gives us a very general way to represent an operation as a function type with domain $I \rightarrow A$ (the type of “ I -tuples”) and codomain A . For example, the *I -ary projection operations* on A are represented as inhabitants of the type `Op I A` as follows.

$\pi : \{I : \mathcal{V} \ .\} \{A : \mathcal{U} \ .\} \rightarrow I \rightarrow \text{Op } I A$
 $\pi \ i \ x = x \ i$

Let us review the informal definition of compatibility. Suppose A and I are types and fix $f : \text{Op } I A$ and $R : \text{Rel } A \mathcal{W}$ (an I -ary operation and a binary relation on A , respectively). We say that f and R are *compatible* and we write³⁰ $f \vdash R$ just in case $\forall u v : I \rightarrow A$,

$$\Pi i : I, R (u \ i) (v \ i) \rightarrow R (f \ u) (f \ v).$$

To implement this in Agda, we first define a function `eval-rel` which “lifts” a binary relation to the corresponding I -ary relation, and we use this to define the function `|` representing *compatibility of an I -ary operation with a binary relation*.

²⁹The definitions here are from the `Agda Standard Library`, translated into `TypeTopo/UALib` notation.

³⁰The symbol `|` denoting compatibility comes from Cliff Bergman’s universal algebra textbook [1].

```

eval-rel : {A :  $\mathcal{U}$  ·} {I :  $\mathcal{V}$  ·} → Rel A  $\mathcal{W}$  → Rel (I → A) ( $\mathcal{V} \sqcup \mathcal{W}$ )
eval-rel R u v =  $\Pi$  i :  $\_$  , R (u i) (v i)

 $\_$ ! : {A :  $\mathcal{U}$  ·} {I :  $\mathcal{V}$  ·} → Op I A → Rel A  $\mathcal{W}$  →  $\mathcal{U} \sqcup \mathcal{V} \sqcup \mathcal{W}$  ·
f! : R = (eval-rel R) = [ f ]⇒ R

```

3.2 Continuous relations: arbitrary-sorted relations of arbitrary arity^{*31}

This section presents the `Relations.Continuous` module of the `AgdaUALib`.

3.2.1 Motivation

In set theory, an n -ary relation on a set A is a subset of the n -fold product $A \times \cdots \times A$. We could try to model these as predicates over a product type representing $A \times \cdots \times A$, or as relations of type $A \rightarrow A \rightarrow \cdots \rightarrow A \rightarrow \mathcal{W}$ · (for some universe \mathcal{W}). To implement such types requires knowing the arity in advance, and then form an n -fold product or n -fold arrow. It turns out to be both easier and more general if we define an arity type $I : \mathcal{V}$ ·, and define the type representing I -ary relations on A as the function type $(I \rightarrow A) \rightarrow \mathcal{W}$ ·. Then, if we happen to be interested in n -ary relations for some natural number n , we could take I to be the n -element type `Fin n`, [23].

Below we will define `ContRel` to be the type $(I \rightarrow A) \rightarrow \mathcal{W}$ · and we will call this the type of *continuous relations*. This generalizes the discrete relations we defined in `Relations.Discrete` (unary, binary, etc.) since continuous relations can be of arbitrary arity. They are not completely general, however, since they are defined over a single type. Said another way, these are “single-sorted” relations. We will remove this limitation when we define the type of *dependent continuous relations*. Just as `Rel A \mathcal{W}` was the single-sorted special case of the multisorted `REL A B \mathcal{W}` type, so too will `ContRel I A \mathcal{W}` be the single-sorted version of dependent continuous relations. The latter will represent relations that not only have arbitrary arities, but also are defined over arbitrary families of types.

To be more concrete, given an arbitrary family $A : I \rightarrow \mathcal{U}$ · of types, we may have a relation from $A\ i$ to $A\ j$ to $A\ k$ to \dots , where the collection represented by the indexing type I might not even be enumerable.³² We will refer to such relations as *dependent continuous relations* (or *dependent relations*) because the definition of a type that represents them requires dependent types. The `DepRel` type that we define below manifests this completely general notion of relation.

3.2.2 Continuous relation types

We now define the type `ContRel` which represents predicates of arbitrary arity over a single type A . We call this the type of *continuous relations*.³³

³¹ Sections marked with an asterisk include new types that are more abstract and general (and frankly more interesting) than the ones defined in other sections. Moreover, the types defined in starred sections are used in only a few other places in the `AgdaUALib`, so they may be safely skimmed or even skipped.

³² Because the collection represented by the indexing type I might not even be enumerable, technically speaking, instead of “ $A\ i$ to $A\ j$ to $A\ k$ to \dots ,” we should have written something like “`TO (i : I) , A i`”

³³ For consistency and readability, throughout the `UALib` we reserve two universe variables for special purposes. The first of these is \mathcal{O} which shall be reserved for types that represent *operation symbols* (see `Algebras.Signatures`). The second is \mathcal{V} which we reserve for types representing *arities* of relations or operations.

$\text{ContRel} : \mathcal{V} \rightarrow \mathcal{U} \rightarrow (\mathcal{W} : \text{Universe}) \rightarrow \mathcal{V} \sqcup \mathcal{U} \sqcup \mathcal{W}^+ \cdot$
 $\text{ContRel } I A \mathcal{W} = (I \rightarrow A) \rightarrow \mathcal{W} \cdot$

Next we present types that are useful for asserting and proving facts about the compatibility of functions with continuous relations. The first is an *evaluation* function which “lifts” an I -ary relation to an $(I \rightarrow J)$ -ary relation. The lifted relation will relate an I -tuple of J -tuples when the “ I -slices” (or “rows”) of the J -tuples belong to the original relation. The second definition is a type that represents compatibility of an operation with a continuous relation. (We’ll dissect these definitions below.)

$\text{eval-cont-rel} : \text{ContRel } I A \mathcal{W} \rightarrow (I \rightarrow J \rightarrow A) \rightarrow \mathcal{V} \sqcup \mathcal{W} \cdot$
 $\text{eval-cont-rel } R a = \Pi j : J, R \lambda i \rightarrow a i j$

$\text{cont-compatible-op} : \text{Op } J A \rightarrow \text{ContRel } I A \mathcal{W} \rightarrow \mathcal{V} \sqcup \mathcal{U} \sqcup \mathcal{W} \cdot$
 $\text{cont-compatible-op } f R = \Pi a : (I \rightarrow J \rightarrow A), (\text{eval-cont-rel } R a \rightarrow R \lambda i \rightarrow (f (a i)))$

To readers who find the syntax of the last two definitions nauseating, we recommend focusing on the semantics. First, internalize the fact that $a : I \rightarrow J \rightarrow A$ denotes an I -tuple of J -tuples of inhabitants of A . Next, recall that a continuous relation R represents a certain collection of I -tuples. Specifically, if $x : I \rightarrow A$ is an I -tuple, then $R x$ denotes the assertion that “ x belongs to R ” or “ x satisfies R .” For each continuous relation R , the type $\text{eval-cont-rel } R$ represents a certain collection of I -tuples of J -tuples, namely, the tuples $a : I \rightarrow J \rightarrow A$ for which $\text{eval-cont-rel } R a$ holds. For simplicity, pretend that J is a finite set, say, $\{1, 2, \dots, J\}$, so that we can write down a couple of the J -tuples as columns. For example, here are the i -th and k -th columns (for some $i k : I$).

$$\begin{array}{cc}
 a i 1 & a k 1 \\
 a i 2 & a k 2 \\
 a i 3 & a k 3 \\
 \vdots & \vdots \\
 a i J & a k J
 \end{array} \leftarrow (\text{if there are } I \text{ columns, then each row forms an } I\text{-tuple})$$

Now $\text{eval-cont-rel } R a$ is defined by $\forall j \rightarrow R (\lambda i \rightarrow (a i) j)$ which represents the assertion that each row of the I columns shown above belongs to the original relation R . Finally, $\text{cont-compatible-op}$ takes a J -ary operation on A , say, $f : \text{Op } J A$, and an I -tuple $a i : J \rightarrow A$ of J -tuples, and determines whether the I -tuple $\lambda i \rightarrow f (a i)$ belongs to R .

3.2.3 Dependent relations

In this section we exploit the power of dependent types to define a completely general relation type. Specifically, we let the tuples inhabit a dependent function type, where the codomain may depend upon the input coordinate $i : I$ of the domain. Heuristically, think of the inhabitants of the following type as relations from $A i$ to $A j$ to $A k$ to \dots (This is only an heuristic since I can represent an uncountable collection.³²)

$\text{DepRel} : (I : \mathcal{V} \rightarrow \mathcal{U}) \rightarrow (I \rightarrow \mathcal{U} \rightarrow \mathcal{W} : \text{Universe}) \rightarrow \mathcal{V} \sqcup \mathcal{U} \sqcup \mathcal{W}^+ \cdot$
 $\text{DepRel } I \mathcal{A} \mathcal{W} = \Pi \mathcal{A} \rightarrow \mathcal{W} \cdot$

We call DepRel the type of *dependent relations*.

Above we saw lifts of continuous relations and what it means for such relations to be compatible with functions. We conclude this module by defining the (only slightly more complicated)

lift of dependent relations, and the type that represents compatibility of a tuple of operations with a dependent relation. Here we assume $I J : \mathcal{V} \cdot$ and $\mathcal{A} : I \rightarrow \mathcal{U} \cdot$.

```
eval-dep-rel : DepRel I  $\mathcal{A}$   $\mathcal{W}$   $\rightarrow (\forall i \rightarrow (J \rightarrow \mathcal{A} i)) \rightarrow \mathcal{V} \sqcup \mathcal{W} \cdot$ 
eval-dep-rel R  $\alpha = \forall j \rightarrow R (\lambda i \rightarrow (\alpha i) j)$ 

dep-compatible-op :  $(\forall i \rightarrow \text{Op } J (\mathcal{A} i)) \rightarrow \text{DepRel } I \mathcal{A} \mathcal{W} \rightarrow \mathcal{V} \sqcup \mathcal{U} \sqcup \mathcal{W} \cdot$ 
dep-compatible-op f R =  $\forall \alpha \rightarrow (\text{eval-dep-rel } R) \alpha \rightarrow R \lambda i \rightarrow (f i)(\alpha i)$ 
```

where we let Agda infer that the type of α is $\Pi i : I, (J \rightarrow \mathcal{A} i)$.

3.3 Quotients: equivalences, class representatives, quotient types

This section presents the `Relations.Quotients` module of the `AgdaUALib`, slightly abridged.³⁴

3.3.1 Properties of binary relations

In the `Relations.Discrete` module we defined types for representing and reasoning about binary relations on A . In this module we will define types for binary relations that have special properties. The most important special properties of relations are the ones we now define.

```
Refl :  $\{A : \mathcal{U} \cdot\} \rightarrow \text{Rel } A \mathcal{W} \rightarrow \mathcal{U} \sqcup \mathcal{W} \cdot$ 
Refl  $\_ \approx \_ = \forall \{x\} \rightarrow x \approx x$ 

Symm :  $\{A : \mathcal{U} \cdot\} \rightarrow \text{Rel } A \mathcal{W} \rightarrow \mathcal{U} \sqcup \mathcal{W} \cdot$ 
Symm  $\_ \approx \_ = \forall \{x\} \{y\} \rightarrow x \approx y \rightarrow y \approx x$ 

Antisymm :  $\{A : \mathcal{U} \cdot\} \rightarrow \text{Rel } A \mathcal{W} \rightarrow \mathcal{U} \sqcup \mathcal{W} \cdot$ 
Antisymm  $\_ \approx \_ = \forall \{x\} \{y\} \rightarrow x \approx y \rightarrow y \approx x \rightarrow x \equiv y$ 

Trans :  $\{A : \mathcal{U} \cdot\} \rightarrow \text{Rel } A \mathcal{W} \rightarrow \mathcal{U} \sqcup \mathcal{W} \cdot$ 
Trans  $\_ \approx \_ = \forall \{x\} \{y\} \{z\} \rightarrow x \approx y \rightarrow y \approx z \rightarrow x \approx z$ 
```

Here is a type from `TypeTopo` that expresses *proof-irrelevance* for binary relations.

```
is-subsingleton-valued :  $\{A : \mathcal{U} \cdot\} \rightarrow \text{Rel } A \mathcal{W} \rightarrow \mathcal{U} \sqcup \mathcal{W} \cdot$ 
is-subsingleton-valued  $\_ \approx \_ = \forall x y \rightarrow \text{is-subsingleton } (x \approx y)$ 
```

Thus, if $R : \text{Rel } A \mathcal{W}$, then `is-subsingleton-valued` R asserts that for each pair $x y : A$ there is *at most one proof* of $R x y$. In the section on *Truncation* below (§ 3.4) we introduce a number of similar but more general types to represent *uniqueness-of-proofs* principles for relations of arbitrary arity, over arbitrary types.

3.3.2 Equivalence relations

A binary relation is called a *preorder* if it is reflexive and transitive. An *equivalence relation* is a symmetric preorder. Here are the types we use to represent these concepts in the `UALib`.

```
record IsPreorder  $\{A : \mathcal{U} \cdot\} (\_ \approx \_ : \text{Rel } A \mathcal{W}) : \mathcal{U} \sqcup \mathcal{W} \cdot$  where
  field rfl : Refl  $\_ \approx \_ ;$  trans : Trans  $\_ \approx \_$ 
```

³⁴For unabridged docs and source code see <https://ualib.gitlab.io/Relations.Quotients.html> and <https://gitlab.com/ualib/ualib.gitlab.io/-/blob/master/UALib/Relations/Quotients.lagda>.

```

Preorder :  $\mathcal{U} \rightarrow \mathcal{U} \sqcup \mathcal{W}^+ \rightarrow \cdot$ 
Preorder A =  $\Sigma R : \text{Rel } A \mathcal{W}, \text{IsPreorder } R$ 

record IsEquivalence {A :  $\mathcal{U}$ } {B :  $\mathcal{W}$ } (f : A  $\rightarrow$  B) :  $\mathcal{U} \sqcup \mathcal{W} \rightarrow \cdot$  where
  field rfl : Refl  $\_ \approx \_;$  sym : Symm  $\_ \approx \_;$  trans : Trans  $\_ \approx \_$ 

Equivalence :  $\mathcal{U} \rightarrow \mathcal{U} \sqcup \mathcal{W}^+ \rightarrow \cdot$ 
Equivalence A =  $\Sigma R : \text{Rel } A \mathcal{W}, \text{IsEquivalence } R$ 

```

An easy first example of an equivalence relation is the kernel of any function. Here is how we prove that the kernel of a function is, indeed, an equivalence relation on the domain of the function.

```

ker-IsEquivalence : {A :  $\mathcal{U}$ } {B :  $\mathcal{W}$ } (f : A  $\rightarrow$  B)  $\rightarrow$  IsEquivalence (ker f)
ker-IsEquivalence f = record { rfl = refl; sym =  $\lambda z \rightarrow \equiv\text{-sym } z;$  trans =  $\lambda p \ q \rightarrow \equiv\text{-trans } p \ q$  }

```

3.3.3 Truncated preorders and equivalences

Using the `is-subsingleton-valued` type defined earlier, we can define the type of preorders and equivalences that have “unique identity proofs” as follows.

```

record IsPreord {A :  $\mathcal{U}$ } (R : Rel A  $\mathcal{W}$ ) :  $\mathcal{U} \sqcup \mathcal{W} \rightarrow \cdot$  where
  field is-preorder : IsPreorder R
  is-truncated : is-subsingleton-valued R

Preord :  $\mathcal{U} \rightarrow \mathcal{U} \sqcup \mathcal{W}^+ \rightarrow \cdot$ 
Preord A =  $\Sigma R : \text{Rel } A \mathcal{W}, \text{IsPreord } R$ 

record IsEqv {A :  $\mathcal{U}$ } (R : Rel A  $\mathcal{W}$ ) :  $\mathcal{U} \sqcup \mathcal{W} \rightarrow \cdot$  where
  field is-equivalence : IsEquivalence  $\_ \approx \_$ 
  is-truncated : is-subsingleton-valued  $\_ \approx \_$ 

Eqv :  $\mathcal{U} \rightarrow \mathcal{U} \sqcup \mathcal{W}^+ \rightarrow \cdot$ 
Eqv A =  $\Sigma R : \text{Rel } A \mathcal{W}, \text{IsEqv } R$ 

```

3.3.4 Equivalence classes (blocks)

Let R be an equivalence relation on A . Then for each $a : A$ there is an *equivalence class* that contains a which we denote and define by $[a] R := \{x : A \mid R \ x \ a\}$. We call this the *R-block containing a*.

```

[ ] : {A :  $\mathcal{U}$ }  $\rightarrow$  A  $\rightarrow$  Rel A  $\mathcal{W} \rightarrow$  Pred A  $\mathcal{W}$ 
[a] R =  $\lambda x \rightarrow R \ a \ x$ 

```

Thus, $x \in [a] R$ if and only if $R \ a \ x$. The members of the R -block containing a as a predicate C satisfying `IsBlock C`, where the latter is defined as follows.

```

IsBlock : {A :  $\mathcal{U}$ } {C : Pred A  $\mathcal{W}}$  {R : Rel A  $\mathcal{W}}$   $\rightarrow$   $\mathcal{U} \sqcup \mathcal{W}^+ \rightarrow \cdot$ 
IsBlock {A} C {R} =  $\Sigma a : A, C \equiv [a] R$ 

```

That is, C is the R -block containing a provided there exists a pair $(a, p) : \text{IsBlock } R \ C$ where p is a proof of $C \equiv [a] R$.

If R is an equivalence relation on A , then the *quotient* of A modulo R is denoted by A / R and defined as the collection $\{[a] R \mid a : A\}$ of all R -blocks of A . We represent the quotient by the following type.

```

_/_ : (A :  $\mathcal{U}$   $\cdot$ ) → Rel A  $\mathcal{W}$  →  $\mathcal{U} \sqcup (\mathcal{W}^+)$   $\cdot$ 
A / R =  $\Sigma$  C : Pred A  $\mathcal{W}$  , IsBlock C {R}

```

We use the following type to represent an R -block with a designated representative.

```

⟨⟦_⟧⟩ : {A :  $\mathcal{U}$   $\cdot$ } → A → {R : Rel A  $\mathcal{W}$ } → A / R
⟨⟦ a ⟧⟩ {R} = [ a ] R , (a , refl)

```

This serves as a kind of *introduction rule*. Dually, the next type provides an *elimination rule*.³⁵

```

⊢_/_ : {A :  $\mathcal{U}$   $\cdot$ } {R : Rel A  $\mathcal{W}$ } → A / R → A
⊢ C , (a , p) ⊢ = a

```

Later we will need the following tools for working with the quotient types defined above.

```

module _ { $\mathcal{U}$   $\mathcal{W}$  : Universe} {A :  $\mathcal{U}$   $\cdot$ } {x y : A} {R : Rel A  $\mathcal{W}$ } where

  open IsEquivalence
  /-subset : IsEquivalence R → R x y → [ x ] R  $\subseteq$  [ y ] R
  /-subset Req Rxy {z} Rxz = (trans Req) ((sym Req) Rxy) Rxz

  /-supset : IsEquivalence R → R x y → [ y ] R  $\subseteq$  [ x ] R
  /-supset Req Rxy {z} Ryz = (trans Req) Rxy Ryz

  /-≐ : IsEquivalence R → R x y → [ x ] R  $\doteq$  [ y ] R
  /-≐ Req Rxy = /-subset Req Rxy , /-supset Req Rxy

```

3.4 Truncation: continuous propositions, quotient extensionality

This section presents the `Relations.Truncation` module of the `AgdaUALib`, slightly abridged.³⁶ Here we discuss *truncation* and *h-sets* (which we just call *sets*). We first give a brief discussion of standard notions of *truncation* from a viewpoint that seems useful for formalizing mathematics in Agda.³⁷

3.4.1 Background and motivation

This brief introduction to *truncation* is intended for novices. Those already familiar with the concept might wish to skip to the next subsection.

In general, we may have multiple inhabitants of a given type, hence (via Curry-Howard) multiple proofs of a given proposition. For instance, suppose we have a type X and an identity relation $_ \equiv_0 _$ on X so that, given two inhabitants of X , say, $a b : X$, we can form the type $a \equiv_0 b$. Suppose p and q inhabit the type $a \equiv_0 b$; that is, p and q are proofs of $a \equiv_0 b$, in which case we write $p \ q : a \equiv_0 b$. We might then wonder whether the two proofs p and q are equivalent.

³⁵ **Unicode Hint.** Type `⌈` and `⌊` as `\cul` and `\cur` in `agda2-mode`.

³⁶ For unabridged docs and source code see <https://ualib.gitlab.io/Relations.Truncation.html> and <https://gitlab.com/ualib/ualib.gitlab.io/-/blob/master/UALib/Relations/Truncation.lagda>.

³⁷ Readers wishing to learn more about truncation may wish to consult [10, §34], [3], or [18, §7.1].

We are asking about an identity type on the identity type \equiv_0 , and whether there is some inhabitant, say, r of this type; i.e., whether there is a proof $r : p \equiv_1 q$ that the proofs of $a \equiv_0 b$ are the same. If such a proof exists for all $p q : a \equiv_0 b$, then the proof of $a \equiv_0 b$ is unique; as a property of the types X and \equiv_0 , this is sometimes called *uniqueness of identity proofs*.

Now, perhaps we have two proofs, say, $r s : p \equiv_1 q$ that the proofs p and q are equivalent. Then of course we wonder whether $r \equiv_2 s$ has a proof! But at some level we may decide that the potential to distinguish two proofs of an identity in a meaningful way (so-called *proof-relevance*) is not useful or desirable. At that point, say, at level k , we would be naturally inclined to assume that there is at most one proof of any identity of the form $p \equiv_k q$. This is called **truncation** (at level k).

3.4.2 Sets

In **homotopy type theory**, a type X with an identity relation \equiv_0 is called a *set* (or *0-groupoid*) if for every pair $x y : X$ there is at most one proof of $x \equiv_0 y$. In other words, the type X , along with its equality type \equiv_x , form a *set* if for all $x y : X$ there is at most one proof of $x \equiv_0 y$.

This notion is formalized in **TypeTopo** using the type **is-set** which is defined using the **is-subsingleton** type (§2.4) as follows.

```
is-set :  $\mathcal{U} \rightarrow \mathcal{U}$ 
is-set A = (x y : A) → is-subsingleton (x ≡ y)
```

Thus, the pair (X, \equiv_0) forms a set iff it satisfies $\forall x y : X \rightarrow \text{is-subsingleton } (x \equiv_0 y)$.

We will also need the function **to- Σ - \equiv** , which is part of Escardó's characterization of *equality in Sigma types*.³⁸ It is defined as follows.

```
to- $\Sigma$ - $\equiv$  : {A :  $\mathcal{U}$ } {B : A →  $\mathcal{W}$ } { $\sigma \tau$  :  $\Sigma B$ }
→  $\Sigma p : | \sigma | \equiv | \tau | , (\text{transport } B p \parallel \sigma \parallel) \equiv \parallel \tau \parallel$ 
→  $\sigma \equiv \tau$ 

to- $\Sigma$ - $\equiv$  (refl {x = x} , refl {x = a}) = refl {x = (x , a)}
```

We will use **is-embedding**, **is-set**, and **to- Σ - \equiv** in the next subsection to prove that a monic function into a set is an embedding.

3.4.3 Injective functions are set embeddings

Before moving on to define propositions, we discharge an obligation mentioned but left unfulfilled in the **embeddings** section of the **Overture.Inverses** module. Recall, we described and imported the **is-embedding** type, and we remarked that an embedding is not simply a monic function. However, if we assume that the codomain is truncated so as to have unique identity proofs, then we can prove that every monic function into that codomain will be an embedding. On the other hand, embeddings are always monic, so we will end up with an equivalence.

```
monic-is-embedding[sets] : (f : A → B) → is-set B → Monic f → is-embedding f

monic-is-embedding[sets] f Bset fmon b (u , fu≡b) (v , fv≡b) =  $\gamma$ 
where
```

³⁸See ([10]), specifically,
<https://www.cs.bham.ac.uk/~mhe/HoTT-UF-in-Agda-Lecture-Notes/HoTT-UF-Agda.html#sigmaequality>

```

fuv : f u ≡ f v
fuv = ≡-trans fu≡b (fv≡b -1)

uv : u ≡ v
uv = fmon u v fuv

arg1 : ∑ p : (u ≡ v) , (transport (λ a → f a ≡ b) p fu≡b) ≡ fv≡b
arg1 = uv , Bset (f v) b (transport (λ a → f a ≡ b) uv fu≡b) fv≡b

γ : u , fu≡b ≡ v , fv≡b
γ = to-Σ-≡ arg1

```

In stating the previous result, we introduce a new convention to which we will try to adhere. If the antecedent of a theorem includes the assumption that one of the types involved is a set, then we add to the name of the theorem the suffix `|sets`, which calls to mind the standard mathematical notation for the restriction of a function to a subset of its domain.

Embeddings are always monic, so we conclude that when a function’s codomain is a set, then that function is an embedding if and only if it is monic.

```

embedding-iff-monic|sets : (f : A → B) → is-set B → is-embedding f ⇔ Monic f
embedding-iff-monic|sets f Bset = (embedding-is-monic f) , (monic-is-embedding|sets f Bset)

```

3.4.4 Propositions

Sometimes we will want to assume that a type A is a *set*. As we just learned, this means there is at most one proof that two inhabitants of A are the same. Analogously, for predicates on A , we may wish to assume that there is at most one proof that an inhabitant of X satisfies the given predicate. If a unary predicate satisfies this condition, then we call it a *unary proposition*. We now define a type that captures this concept.³⁹

```

Pred1 :  $\mathcal{U} \rightarrow (\mathcal{W} : \mathbf{Universe}) \rightarrow \mathcal{U} \sqcup \mathcal{W}^+ \rightarrow \mathbf{Prop}$ 
Pred1 A  $\mathcal{W}$  = ∑ P : (Pred A  $\mathcal{W}$ ) , ∀ x → is-subsingleton (P x)

```

The principle of *proposition extensionality* asserts that logically equivalent propositions are equivalent. That is, if we have $P Q : \text{Pred}_1$ and $| P | \subseteq | Q |$ and $| Q | \subseteq | P |$, then $P \equiv Q$. This is formalized as follows.⁴⁰

```

prop-ext : ( $\mathcal{U} \mathcal{W} : \mathbf{Universe}$ ) → ( $\mathcal{U} \sqcup \mathcal{W}$ )+ →  $\mathbf{Prop}$ 
prop-ext  $\mathcal{U} \mathcal{W}$  = ∀ {A :  $\mathcal{U}$  } {P Q : Pred1 A  $\mathcal{W}$ } → | P | ⊆ | Q | → | Q | ⊆ | P | → P ≡ Q

```

Recall, we defined the relation $\dot{=}$ for predicates as follows: $P \dot{=} Q = (P \subseteq Q) \times (Q \subseteq P)$. Therefore, if we postulate `prop-ext` $\mathcal{U} \mathcal{W}$ and $P \dot{=} Q$, then $P \equiv Q$ obviously follows. Nonetheless, let us record this observation, assuming the context includes $\{\mathcal{U} \mathcal{W} : \mathbf{Universe}\} \{A : \mathcal{U} \}$.

```

prop-ext' : prop-ext  $\mathcal{U} \mathcal{W}$  → {P Q : Pred1 A  $\mathcal{W}$ } → | P | ≐ | Q | → P ≡ Q
prop-ext' pe hyp = pe (fst hyp) (snd hyp)

```

³⁹ Agda now has a type called `Prop`, though we have never tried to use it. It likely provides at least some of the functionality we develop here, however, our preference is to assume only a minimal `MLTT` foundation and build up the types we need ourselves. For details about `Prop`, consult the official documentation at <https://agda.readthedocs.io/en/v2.6.1.3/language/prop.html>.

⁴⁰ cf. [10], specifically the section on “Prop extensionality and the powerset”.

The foregoing easily generalizes to binary relations. If R is a binary relation such that there is at most one way to prove that a given pair of elements is R -related, then we call R a *binary proposition*. As above, we use the `is-subsingleton` type of `TypeTopo` to impose this truncation assumption on a binary relation.⁴¹

```
Pred2 : {U : Universe} → U → (W : Universe) → U → W → •
Pred2 A W = Σ R : (Rel A W) , is-subsingleton-valued R
```

Recall, `is-subsingleton-valued` is simply defined as

```
is-subsingleton-valued R = ∀ x y → is-subsingleton (R x y)
```

which is the assertion that for all $x y$ there is at most one proof that x and y are R -related. We will generalize this from binary to arbitrary (i.e., continuous and dependent) relations below (see `IsContProp` and `IsDepProp`).

3.4.5 Quotient extensionality

We need a (subsingleton) identity type for congruence classes over sets so that we can equate two classes even when they are presented using different representatives. Proposition extensionality is precisely what we need to accomplish this. We now define a type called `class-extensionality` that will play a crucial role later (e.g., in the formal proof of Birkhoff's HSP theorem).⁴²

```
module _ {U W : Universe} {A : U → W} {R : Rel A W} where
  open IsEqv
  class-extensionality : prop-ext U W → IsEqv R → {u v : A} → R u v → [ u ] R ≡ [ v ] R
  class-extensionality pe Req {u}{v} Ruv = ap fst PQ
  where
    P Q : Pred1 A W
    P = (λ a → R u a) , (λ a → is-truncated Req u a)
    Q = (λ a → R v a) , (λ a → is-truncated Req v a)

    α : [ u ] R ⊆ [ v ] R
    α ua = fst (/≐ (is-equivalence Req) Ruv) ua

    β : [ v ] R ⊆ [ u ] R
    β va = snd (/≐ (is-equivalence Req) Ruv) va

    PQ : P ≡ Q
    PQ = (prop-ext' pe (α , β))

  to-subtype-⟦ ⟧ : (∀ C → is-subsingleton (IsBlock R C))
    → {C D : Pred A W} {c : IsBlock R C} {d : IsBlock R D}
    → C ≡ D → (C , c) ≡ (D , d)

  to-subtype-⟦ ⟧ ssA {C}{D}{c}{d} CD = to-Σ≡ (CD , ssA D (transport (IsBlock R) CD c) d)
```

⁴¹This is another example of *proof-irrelevance*. Indeed, if R is a binary proposition and we have two proofs of $R x y$, then we can assume that the proofs are indistinguishable or that any distinctions are irrelevant.

⁴²Previous proofs of the `class-extensionality` theorem required *function extensionality* (§2.3); however, as the proof given here makes clear, this is unnecessary.

```

class-extensionality' : prop-ext  $\mathcal{U} \mathcal{W} \rightarrow (\forall C \rightarrow \text{is-subsingleton } (\text{IsBlock } R C))$ 
→
    IsEqv  $R \rightarrow \{u v : A\} \rightarrow R u v \rightarrow \langle\langle u \rangle\rangle \equiv \langle\langle v \rangle\rangle$ 

class-extensionality' pe ssA Reqv Ruw = to-subtype- $\langle\langle \rangle\rangle$  ssA (class-extensionality pe Reqv Ruw)

```

3.4.6 General propositions*³¹

In this final subsection of our presentation of relations in type theory, we offer a few interesting new types to complement the types we defined in the module [Relations.Continuous](#) (§3.2). We should point out however that (so far) no other modules of the library depend on the types defined here. Therefore, the reader may safely skip to Section 4 without fear that this will lead to confusion at some point later on.

We defined a type called [ContRel](#) (in §3.2.2) to represent relations of arbitrary arity. Here we introduce a new type of *truncated continuous relations*, the inhabitants of which we call *continuous propositions*. (Assume the context includes $\mathcal{U} : \text{Universe}$ and $I : \mathcal{V} \cdot$.)

```

IsContProp : {A :  $\mathcal{U} \cdot$ }{ $\mathcal{W} : \text{Universe}$ } → ContRel  $I A \mathcal{W} \rightarrow \mathcal{V} \sqcup \mathcal{U} \sqcup \mathcal{W} \cdot$ 
IsContProp {A = A} P =  $\Pi a : (I \rightarrow A)$  , is-subsingleton (P a)

ContProp :  $\mathcal{U} \cdot \rightarrow (\mathcal{W} : \text{Universe}) \rightarrow \mathcal{U} \sqcup \mathcal{V} \sqcup \mathcal{W}^+ \cdot$ 
ContProp A  $\mathcal{W} = \Sigma P : (\text{ContRel } I A \mathcal{W})$  , IsContProp P

cont-prop-ext :  $\mathcal{U} \cdot \rightarrow (\mathcal{W} : \text{Universe}) \rightarrow \mathcal{U} \sqcup \mathcal{V} \sqcup \mathcal{W}^+ \cdot$ 
cont-prop-ext A  $\mathcal{W} = \{P Q : \text{ContProp } A \mathcal{W}\} \rightarrow | P | \subseteq | Q | \rightarrow | Q | \subseteq | P | \rightarrow P \equiv Q$ 

```

To see the point of the types just defined, suppose [cont-prop-ext](#) $A \mathcal{W}$ holds. Then we can prove that logically equivalent continuous propositions of type [ContProp](#) $A \mathcal{W}$ are equivalent. In other words, under the stated hypotheses, we obtain the following extensionality lemma for continuous propositions.

```

cont-prop-ext' : {A :  $\mathcal{U} \cdot$ }{ $\mathcal{W} : \text{Universe}$ } → cont-prop-ext A  $\mathcal{W}$ 
→
    {P Q : ContProp A  $\mathcal{W}$ } → | P |  $\dot{=}$  | Q | → P  $\equiv$  Q
cont-prop-ext' pe hyp = pe | hyp | || hyp ||

```

While we're at it, we might as well take the abstraction one step further and define the type of *truncated dependent relations*, which we call *dependent propositions*.

```

IsDepProp : {I :  $\mathcal{V} \cdot$ }{ $\mathcal{A} : I \rightarrow \mathcal{U} \cdot$ }{ $\mathcal{W} : \text{Universe}$ } → DepRel  $I \mathcal{A} \mathcal{W} \rightarrow \mathcal{V} \sqcup \mathcal{U} \sqcup \mathcal{W} \cdot$ 
IsDepProp {I = I} { $\mathcal{A}$ } P =  $\Pi a : \Pi \mathcal{A}$  , is-subsingleton (P a)

DepProp : (I →  $\mathcal{U} \cdot$ ) → ( $\mathcal{W} : \text{Universe}$ ) →  $\mathcal{U} \sqcup \mathcal{V} \sqcup \mathcal{W}^+ \cdot$ 
DepProp  $\mathcal{A} \mathcal{W} = \Sigma P : (\text{DepRel } I \mathcal{A} \mathcal{W})$  , IsDepProp P

dep-prop-ext : (I →  $\mathcal{U} \cdot$ ) → ( $\mathcal{W} : \text{Universe}$ ) →  $\mathcal{U} \sqcup \mathcal{V} \sqcup \mathcal{W}^+ \cdot$ 
dep-prop-ext  $\mathcal{A} \mathcal{W} = \{P Q : \text{DepProp } \mathcal{A} \mathcal{W}\} \rightarrow | P | \subseteq | Q | \rightarrow | Q | \subseteq | P | \rightarrow P \equiv Q$ 

```

4 Algebra Types

Finally we are ready to present the [Algebras](#) module of the [AgdaUALib](#). Here we use type theory and Agda to codify the most basic objects of universal algebra, such as *operations* and *signatures* (§4.1), *algebras* (§4.2), *product algebras* (§4.3), *congruence relations* and *quotient algebras* (§4.4).

A popular way to represent algebraic structures in type theory is with record types. The Sigma type (§2.1.4) provides an equivalent alternative that we happen to prefer and we use it throughout the library, both for consistency and because of its direct connection to the existential quantifier of logic. Recall from §2.1.4 that the type $\Sigma x : X, P x$ represents the proposition, “there exists x in X such that $P x$ holds;” in symbols, $\exists x \in X, P x$. Indeed, an inhabitant of $\Sigma x : X, P x$ is a pair (x, p) such that x inhabits X and p is a proof of $P x$. In other terms, the pair (x, p) is a witness and proof of the proposition $\exists x \in X, P x$.

4.1 Signatures: types for operations & signatures

This section presents the `Algebras.Signatures` module of the `AgdaUALib`, slightly abridged.⁴³

4.1.1 Operation type

We begin by defining the type of *operations* as follows.

```
Op :  $\mathcal{V} \rightarrow \mathcal{U} \rightarrow \mathcal{U} \sqcup \mathcal{V}$ 
Op I A = (I  $\rightarrow$  A)  $\rightarrow$  A
```

The type `Op` encodes the arity of an operation as an arbitrary type $I : \mathcal{V}$, which gives us a very general way to represent an operation as a function type with domain $I \rightarrow A$ (the type of “tuples”) and codomain A . For example, the *I-ary projection operations* on A can be codified as inhabitants of the type `Op I A` in the following way.

```
 $\pi : \{I : \mathcal{V}\} \{A : \mathcal{U}\} \rightarrow I \rightarrow \text{Op } I A$ 
 $\pi \ i \ x = x \ i$ 
```

4.1.2 Signature type

We define the signature of an algebraic structure in Agda like this.

```
Signature : ( $\mathbb{O} \ \mathcal{V} : \text{Universe}$ )  $\rightarrow$  ( $\mathbb{O} \sqcup \mathcal{V}$ )+
Signature  $\mathbb{O} \ \mathcal{V} = \Sigma F : \mathbb{O}, (F \rightarrow \mathcal{V})$ 
```

As mentioned in §3.2, in the `UALib` the symbol \mathbb{O} always denotes the universe of *operation symbol* types, while \mathcal{V} is always the universe of *arity* types.

In §2.1.4 we defined special syntax for the first and second projections—namely, `|_` and `||_||`, respectively. Consequently, if $S : \text{Signature } \mathbb{O} \ \mathcal{V}$ is a signature, then `| S |` denotes the set of operation symbols, and `|| S ||` denotes the arity function. If $f : | S |$ is an operation symbol in the signature S , then `|| S || f` is the arity of f .

4.1.2.1 Example

Here is how we could define the signature for *monoids* as an inhabitant of the type `Signature $\mathbb{O} \ \mathcal{V}$` .

```
data monoid-op :  $\mathbb{O}$  where
  e : monoid-op
  · : monoid-op
```

⁴³For unabridged docs and source code see <https://ualib.gitlab.io/Algebras.Signatures.html> and <https://gitlab.com/ualib/ualib.gitlab.io/-/blob/master/UALib/Algebras/Signatures.lagda>.

```
monoid-sig : Signature 0 U0
monoid-sig = monoid-op , λ { e → 0; · → 2 }
```

As expected, the signature for a monoid consists of two operation symbols, `e` and `·`, and a function $\lambda \{ e \rightarrow 0; \cdot \rightarrow 2 \}$ which maps `e` to the empty type `0` (since `e` is the nullary identity) and maps `·` to the two element type `2` (since `·` is binary).⁴⁴

4.2 Algebras: types for algebras, operation interpretation & compatibility

This section presents the `Algebras.Algebras` module of the `AgdaUALib`, slightly abridged.⁴⁵

4.2.1 The Algebra type

For a fixed signature $S : \text{Signature } 0 \mathcal{V}$ and universe \mathcal{U} , we define the type of *algebras in the signature S* (or *S -algebras*) and with *domain* (or *carrier* or *universe*) $A : \mathcal{U}$ as follows.

```
Algebra : (U : Universe) (S : Signature 0 V) → 0 ⊔ V ⊔ U+ ·
Algebra U S = Σ A : U · ,                – the domain
                Π f : | S | , Op (|| S || f) A – the basic operations
```

To be precise, we might call an inhabitant of this type a “ ∞ -algebra” because its domain can be an arbitrary type, say, $A : \mathcal{U}$ and need not be truncated at some level. (In particular, A need not be a *set*; see the discussion of sets in §3.4.2.) We could then take this opportunity to define the type of “0-algebras” (algebras whose domains are sets), which may be closer to what most of us think of when doing informal universal algebra. However, we have found that we only need to assume the domains of our algebras are sets in a few places in the `UALib`, so it seems preferable to work with general (∞ -)algebras throughout and then assume *uniqueness of identity proofs* explicitly and only where needed.

4.2.2 Algebras as record types

Some people prefer to represent algebraic structures in type theory using records, and for those folks we offer the following (equivalent) definition.

```
record algebra (U : Universe) (S : Signature 0 V) : (0 ⊔ V ⊔ U+ ·) where
  constructor mkalg
  field
    univ : U ·
    op : (f : | S |) → ((|| S || f) → univ) → univ
```

This representation of algebras as inhabitants of a record type is equivalent to the representation using Sigma types in the sense that a bi-implication between the two representations is obvious.

```
algebra→Algebra : algebra U S → Algebra U S
algebra→Algebra A = (univ A , op A)

Algebra→algebra : Algebra U S → algebra U S
Algebra→algebra A = mkalg | A | || A ||
```

⁴⁴The types `0` and `2` are defined in the `MGS-MLTT` module of `TypeTopo`.

⁴⁵For unabridged docs and source code see <https://ualib.gitlab.io/Algebras.Algebras.html> and <https://gitlab.com/ualib/ualib.gitlab.io/-/blob/master/UALib/Algebras/Algebras.lagda>.

In developing the more advanced modules of the `AgdaUALib`, we have used the Sigma type representation of algebras exclusively, though we occasionally use record types to represent other basic objects (congruences, subalgebras, and others).

4.2.3 Operation interpretation syntax

We now define a convenient shorthand for the interpretation of an operation symbol. This looks more similar to the standard notation one finds in the literature as compared to the double bar notation we started with, so we will use this new notation almost exclusively in the remaining modules of the `UALib`.

$$\begin{aligned} _ \hat{_} &: (f : | S |)(\mathbf{A} : \text{Algebra } \mathcal{U} S) \rightarrow (\| S \| f \rightarrow | \mathbf{A} |) \rightarrow | \mathbf{A} | \\ f \hat{\mathbf{A}} &= \lambda a \rightarrow (\| \mathbf{A} \| f) a \end{aligned}$$

Thus, if $f : | S |$ is an operation symbol in the signature S and if $a : \| S \| f \rightarrow | \mathbf{A} |$ is a tuple of the same arity, then $(f \hat{\mathbf{A}}) a$ denotes the operation f interpreted in \mathbf{A} and evaluated at a .

4.2.4 Lifts of algebras

Recall, in §2.5 we described a common difficulty one encounters when working with a noncumulative universe hierarchy. We made a promise to provide some domain-specific level lifting and lowering methods. Here we fulfill this promise by supplying a couple of bespoke tools designed specifically for our operation and algebra types.

$$\begin{aligned} \text{Lift-op} &: ((I \rightarrow A) \rightarrow A) \rightarrow (\mathcal{W} : \text{Universe}) \rightarrow ((I \rightarrow \text{Lift}\{\mathcal{W}\} A) \rightarrow \text{Lift}\{\mathcal{W}\} A) \\ \text{Lift-op } f \mathcal{W} &= \lambda x \rightarrow \text{lift } (f (\lambda i \rightarrow \text{lower } (x i))) \\ \text{Lift-alg} &: \text{Algebra } \mathcal{U} S \rightarrow (\mathcal{W} : \text{Universe}) \rightarrow \text{Algebra } (\mathcal{U} \sqcup \mathcal{W}) S \\ \text{Lift-alg } \mathbf{A} \mathcal{W} &= \text{Lift } | \mathbf{A} |, (\lambda (f : | S |) \rightarrow \text{Lift-op } (f \hat{\mathbf{A}}) \mathcal{W}) \\ \text{Lift-alg-record-type} &: \text{algebra } \mathcal{U} S \rightarrow (\mathcal{W} : \text{Universe}) \rightarrow \text{algebra } (\mathcal{U} \sqcup \mathcal{W}) S \\ \text{Lift-alg-record-type } \mathbf{A} \mathcal{W} &= \text{mkalg } (\text{Lift } (\text{univ } \mathbf{A})) (\lambda (f : | S |) \rightarrow \text{Lift-op } ((\text{op } \mathbf{A}) f) \mathcal{W}) \end{aligned}$$

What makes the types just defined useful for resolving type level errors is the nice properties they possess. Indeed, the `UALib` contains formal proofs of the following facts.

- `Lift` is a homomorphism.
- `Lift` is an algebraic invariant; that is, an algebra is isomorphic to its lift.
- `Lift` of a subalgebra is again a subalgebra.
- `Lift` preserves identities.

4.2.5 Compatibility of binary relations

We now define the function `compatible` so that, if \mathbf{A} is an algebra and R a binary relation, then `compatible` $\mathbf{A} R$ will denote the assertion that R is *compatible* with all basic operations of \mathbf{A} ; that is, for every operation symbol $f : | S |$ and all pairs $u v : \| S \| f \rightarrow | \mathbf{A} |$ of tuples from the domain of \mathbf{A} , the following implication holds:

$$(\Pi(i : I), R (u i) (v i)) \rightarrow R ((f \hat{\mathbf{A}}) u) ((f \hat{\mathbf{A}}) v).$$

Using the relation $|:$ (defined in § 3.1.4) this implication is expressed compactly as $(f \hat{\mathbf{A}}) | : R$; whence, the following definition.

```
compatible : (A : Algebra U S) → Rel | A | W → O U V W .
compatible A R = ∀ f → (f ^ A) | R
```

4.2.6 Compatibility of continuous relations*³¹

Next we define a type that represents *compatibility of a continuous relation* with all operations of an algebra. First, we define compatibility with a single operation using `cont-compatible-fun` (defined in §3.2.2).

```
cont-compatible-op : | S | → ContRel I | A | W → O U V W .
cont-compatible-op f R = cont-compatible-fun (f ^ A) R
```

Whence, *compatibility of a continuous relation with an algebra* is defined as follows.

```
cont-compatible : ContRel I | A | W → O U V W .
cont-compatible R = Π f : | S | , cont-compatible-op f R
```

4.3 Products: types for products over arbitrary classes

This section presents the `Algebras.Products` module of the `AgdaUALib`, slightly abridged.⁴⁶ We assume a fixed signature $S : \text{Signature } \mathcal{O} \mathcal{V}$ throughout the module by starting with the line `module Algebras.Products {S : Signature O V} where`.

In the `UALib` the *product of S-algebras* is represented by the following type.⁴⁷

```
Π : (A : I → Algebra U S) → Algebra (J U) S

Π A = (Π i : I , | A i |) ,           -- domain of the product algebra
      λ f a i → (f ^ A i) λ x → a x i -- basic operations of the product algebra
```

The type just defined is the one we use whenever the product of an indexed collection of algebras (of type `Algebra`) is required. However, for the sake of completeness, here is how one could define a type representing the product of algebras inhabiting the record type `algebra`.

```
Π' : (A : I → algebra U S) → algebra (J U) S
Π' A = record { univ = ∀ i → univ (A i) ;           -- domain
               op = λ f a i → (op (A i)) f λ x → a x i } -- basic operations
```

Before going further, let us agree on another convenient notational convention, which is used in many of the later modules of the `UALib`. Given a signature $S : \text{Signature } \mathcal{O} \mathcal{V}$, the type `Algebra U S` has universe level $\mathcal{O} \sqcup \mathcal{V} \sqcup \mathcal{U}^+$, and the $\mathcal{O} \sqcup \mathcal{V}$ part remains fixed since \mathcal{O} and \mathcal{V} always denote the universe levels of operation and arity types, respectively. Such levels occur so often in the `UALib` that we define the following shorthand for their universes: $\text{ov } \mathcal{U} := \mathcal{O} \sqcup \mathcal{V} \sqcup \mathcal{U}^+$.

4.3.1 Products of classes of algebras

An arbitrary class \mathcal{K} of algebras is represented as a predicate over the type `Algebra U S`, for some universe level \mathcal{U} and signature S . That is, $\mathcal{K} : \text{Pred}(\text{Algebra } \mathcal{U} S) \mathcal{W}$ for some \mathcal{W} . Later

⁴⁶ For unabridged docs and source code see <https://ualib.gitlab.io/Algebras.Products.html> and <https://gitlab.com/ualib/ualib.gitlab.io/-/blob/master/UALib/Algebras/Products.lagda>.

⁴⁷ Alternative equivalent notation for the domain of the product is $\forall i \rightarrow | A i |$.

we will formally state and prove that the product of all subalgebras of algebras in \mathcal{K} belongs to the class $\mathbf{SP}(\mathcal{K})$ of subalgebras of products of algebras in \mathcal{K} . That is, $\prod \mathbf{S}(\mathcal{K}) \in \mathbf{SP}(\mathcal{K})$. This turns out to be a nontrivial exercise.

To begin, we need to define types that represent products over arbitrary (nonindexed) families such as \mathcal{K} or $\mathbf{S}(\mathcal{K})$. Observe that $\prod \mathcal{K}$ is definitely *not* what we want. To see why, recall that $\mathbf{Pred}(\mathbf{Algebra} \ \mathcal{U} \ S) \ \mathcal{W}$ is just an alias for the function type $\mathbf{Algebra} \ \mathcal{U} \ S \rightarrow \mathcal{W}$. We interpret the latter semantically by taking $\mathcal{K} \ \mathbf{A}$ to be the assertion that $\mathcal{K} \ \mathbf{A}$ belongs to $\mathcal{K} \ \mathbf{A}$, denoted $\mathbf{A} \in \mathcal{K}$. Therefore, by definition, we have

$$\prod \mathcal{K} = \prod \mathbf{A} : (\mathbf{Algebra} \ \mathcal{U} \ S) , \mathcal{K} \ \mathbf{A} = \prod \mathbf{A} : (\mathbf{Algebra} \ \mathcal{U} \ S) , \mathbf{A} \in \mathcal{K}.$$

Semantically, this is the assertion that every algebra of type $\mathbf{Algebra} \ \mathcal{U} \ S$ belongs to \mathcal{K} , and this bears little resemblance to the product of algebras that we seek.

What we need is a type that serves to index the class \mathcal{K} , and a function \mathfrak{A} that maps an index to the inhabitant of \mathcal{K} at that index. But \mathcal{K} is a predicate (of type $(\mathbf{Algebra} \ \mathcal{U} \ S) \rightarrow \mathcal{W}$) and the type $\mathbf{Algebra} \ \mathcal{U} \ S$ seems rather nebulous in that there is no natural indexing class with which to “enumerate” all inhabitants of $\mathbf{Algebra} \ \mathcal{U} \ S$ that belong to \mathcal{K} .⁴⁸

The solution is to essentially take \mathcal{K} itself to be the index type; at least heuristically that is how one can think of the type \mathfrak{I} that we now define.⁴⁹

$$\begin{aligned} \mathfrak{I} &: \mathbf{ov} \ \mathcal{U} \ . \\ \mathfrak{I} &= \sum \mathbf{A} : (\mathbf{Algebra} \ \mathcal{U} \ S) , (\mathbf{A} \in \mathcal{K}) \end{aligned}$$

Taking the product over the index type \mathfrak{I} requires a function that maps an index $i : \mathfrak{I}$ to the corresponding algebra. Each $i : \mathfrak{I}$ denotes a pair, (\mathbf{A} , p) , where \mathbf{A} is an algebra and p is a proof that \mathbf{A} belongs to \mathcal{K} , so the function mapping such an index to the corresponding algebra is simply the first projection.

$$\begin{aligned} \mathfrak{A} &: \mathfrak{I} \rightarrow \mathbf{Algebra} \ \mathcal{U} \ S \\ \mathfrak{A} &= \lambda (i : \mathfrak{I}) \rightarrow | i | \end{aligned}$$

Finally, we represent the product of all members of the class \mathcal{K} by the following type.

$$\begin{aligned} \mathbf{class-product} &: \mathbf{Algebra} \ (\mathbf{ov} \ \mathcal{U}) \ S \\ \mathbf{class-product} &= \prod \mathfrak{A} \end{aligned}$$

Observe that the application $\mathfrak{A} \ (\mathbf{A} , p)$ of \mathfrak{A} to the pair (\mathbf{A} , p) (the result of which is simply the algebra \mathbf{A}) may be viewed as the projection out of the product $\prod \mathfrak{A}$ and onto the “ (\mathbf{A} , p) -th component” of that product.

4.4 Congruences: types for congruences & quotient algebras

This section presents the `Algebras.Congruences` module of the `AgdaUALib`, slightly abridged.⁵⁰ A *congruence relation* of an algebra \mathbf{A} is defined to be an equivalence relation that is compatible with the basic operations of \mathbf{A} . This concept can be represented in a number of alternative but equivalent ways. Informally, a relation is a congruence if and only if it is both an equivalence

⁴⁸ If you haven’t already seen this before, do yourself a favor and give it some thought; see if the correct type comes to you organically.

⁴⁹ **Unicode Hints.** Some of our types are denoted with with Gothic (“mathfrak”) symbols. To produce them in `agda2-mode`, type `\Mf` followed by a letter. For example, `\MfI` \rightsquigarrow \mathfrak{I} .

⁵⁰ For unabridged docs and source code see <https://ualib.gitlab.io/Algebras.Congruences.html> and <https://gitlab.com/ualib/ualib.gitlab.io/-/blob/master/UALib/Algebras/Congruences.lagda>.

relation on the domain of \mathbf{A} and a subalgebra of the square of \mathbf{A} . Formally, we represent a congruence as an inhabitant of either a the Sigma type `Con` or the record type `Congruence`, which we now define.

```

Con : {U : Universe} (A : Algebra U S) → ov U
Con {U} A = Σ θ : ( Rel | A | U ) , IsEquivalence θ × compatible A θ

record Congruence {U W : Universe} (A : Algebra U S) : ov W ⊔ U where
  constructor mkcon
  field
    ⟨_⟩ : Rel | A | W
    Compatible : compatible A ⟨_⟩
    IsEquiv : IsEquivalence ⟨_⟩

```

Each of these types captures the informal notion of congruence, and the only real difference between them is that `Congruence` includes the extra universe parameter \mathcal{W} to accommodate more general underlying relations. Otherwise, the two definitions are equivalent in the sense that each implies the other, as we now prove.

```

Con→Congruence : Con A → Congruence{U} A
Con→Congruence θ = mkcon | θ | (fst || θ ||) (snd || θ ||)

Congruence→Con : Congruence{U} A → Con A
Congruence→Con θ = ⟨ θ ⟩ , IsEquiv θ , Compatible θ

```

We defined the *zero relation* `0-rel` in the `Relations.Discrete` module. We now build the *trivial congruence*, which has `0-rel` as its underlying relation. Observe that `0-rel` is equivalent to the identity relation \equiv and these are obviously both equivalences. In fact, we already proved this of \equiv in the `Overture.Equality` module, so we simply apply the corresponding proofs.

```

0-IsEquivalence : {A : U} → IsEquivalence {A = A} 0
0-IsEquivalence = record {rfl = λ x → refl{x = x}; sym = ≡-symmetric; trans = ≡-transitive}

```

Next we formally record another obvious fact—namely, that `0-rel` is compatible with all operations of all algebras.

```

0-compatible-op : funext V U → {A : Algebra U S} (f : | S |) → compatible-fun (f ^ A) 0
0-compatible-op fe {A} f ptws0 = ap (f ^ A) (fe (λ x → ptws0 x))

```

```

0-compatible : funext V U → {A : Algebra U S} → compatible A 0
0-compatible fe {A} = λ f args → 0-compatible-op fe {A} f args

```

Finally, we have the ingredients need to construct the zero congruence of any algebra we like.

```

Δ : funext V U → {A : Algebra U S} → Congruence A
Δ fe = mkcon 0 (0-compatible fe) 0-IsEquivalence

```

A concrete example is $\llbracket 0 \rrbracket \mathbf{A} \diagdown \theta$, presented in the next subsection.

4.4.1 Quotient Algebras

In many areas of abstract mathematics the *quotient* of an algebra \mathbf{A} with respect to a congruence relation θ of \mathbf{A} plays an important role. This quotient is typically denoted by \mathbf{A} / θ and

Agda allows us to define and express quotients using this standard notation.⁵¹

$$\begin{aligned} _/_ &: (\mathbf{A} : \text{Algebra } \mathcal{U} \ S) \rightarrow \text{Congruence}\{\mathcal{W}\} \ \mathbf{A} \rightarrow \text{Algebra } (\mathcal{U} \sqcup \mathcal{W}^+) \ S \\ \mathbf{A} / \theta &= (\mid \mathbf{A} \mid / \langle \theta \rangle) , & \text{-- the domain of the quotient algebra} \\ \lambda f \ a &\rightarrow \langle (f \hat{\ } \mathbf{A}) (\lambda i \rightarrow \mid \mathbf{a} \ i \mid) \rangle & \text{-- the basic operations of the quotient algebra} \end{aligned}$$

Example. Denote by $\mathbf{0}[\mathbf{A} / \theta]$ the zero relation on the quotient algebra \mathbf{A} / θ , which is defined as follows.

$$\begin{aligned} \mathbf{0}[_/_] &: (\mathbf{A} : \text{Algebra } \mathcal{U} \ S)(\theta : \text{Congruence}\{\mathcal{W}\} \ \mathbf{A}) \rightarrow \text{Rel } (\mid \mathbf{A} \mid / \langle \theta \rangle)(\mathcal{U} \sqcup \mathcal{W}^+) \\ \mathbf{0}[\mathbf{A} / \theta] &= \lambda u \ v \rightarrow u \equiv v \end{aligned}$$

From this we obtain the zero congruence of \mathbf{A} / θ by applying the Δ function defined above.

$$\begin{aligned} \langle \mathbf{0} \rangle _/_ &: (\mathbf{A} : \text{Algebra } \mathcal{U} \ S)(\theta : \text{Congruence}\{\mathcal{W}\} \ \mathbf{A})\{fe : \text{funext } \mathcal{V} \ (\mathcal{U} \sqcup \mathcal{W}^+)\} \\ &\rightarrow \text{Congruence } (\mathbf{A} / \theta) \\ \langle \mathbf{0} \rangle \mathbf{A} / \theta &\{fe\} = \Delta \ fe \end{aligned}$$

Finally, the following *elimination rule* is sometimes useful.

$$\begin{aligned} _/\equiv &: (\theta : \text{Congruence}\{\mathcal{W}\} \ \mathbf{A})\{u \ v : \mid \mathbf{A} \mid\} \rightarrow \langle u \rangle \langle \theta \rangle \equiv \langle v \rangle \rightarrow \langle \theta \rangle u \ v \\ _/\equiv \theta \text{ refl} &= \text{IsEquivalence.rfl } (\text{IsEquiv } \theta) _ \end{aligned}$$

5 Concluding Remarks

We’ve reached the end of Part 1 of our three-part series describing the [AgdaUALib](#). Part 2 will cover homomorphism, terms, and subalgebras, and Part 3 will cover free algebras, equational classes of algebras (i.e., varieties), and Birkhoff’s HSP theorem.

We conclude by noting that one of our goals is to make computer formalization of mathematics more accessible to mathematicians working in universal algebra and model theory. We welcome feedback from the community and are happy to field questions about the [UALib](#), how it is installed, and how it can be used to prove theorems that are not yet part of the library. Merge requests submitted to the UALib’s main gitlab repository are especially welcomed. Please visit the repository at <https://gitlab.com/uallib/uallib.gitlab.io/> and help us improve it.

References

- 1 Clifford Bergman. *Universal Algebra: fundamentals and selected topics*, volume 301 of *Pure and Applied Mathematics (Boca Raton)*. CRC Press, Boca Raton, FL, 2012.
- 2 Ana Bove and Peter Dybjer. *Dependent Types at Work*, pages 57–99. Springer Berlin Heidelberg, Berlin, Heidelberg, 2009. doi:10.1007/978-3-642-03153-3_2.
- 3 Guillaume Brunerie. Truncations and truncated higher inductive types, September 2012. URL: <https://homotopytypetheory.org/2012/09/16/truncations-and-truncated-higher-inductive-types/>.
- 4 Venzio Capretta. Universal algebra in type theory. In *Theorem proving in higher order logics (Nice, 1999)*, volume 1690 of *Lecture Notes in Comput. Sci.*, pages 131–148. Springer, Berlin, 1999. doi:10.1007/3-540-48256-3_10.

⁵¹ **Unicode Hints.** Produce the $_/_$ symbol in `agda2-mode` by typing `\---` and then `C-f` a number of times.

- 5 Jesper Carlström. A constructive version of Birkhoff's theorem. *Mathematical Logic Quarterly*, 54(1):27–34, 2008. doi:[10.1002/malq.200710023](https://doi.org/10.1002/malq.200710023).
- 6 Alonzo Church. A formulation of the simple theory of types. *The Journal of Symbolic Logic*, 5(2):56–68, 1940. URL: <http://www.jstor.org/stable/2266170>.
- 7 Jesper Cockx. Dependent pattern matching and proof-relevant unification, 2017. URL: <https://lirias.kuleuven.be/retrieve/456787>.
- 8 William DeMeo. The Agda Universal Algebra Library, Part 2: Structure. *CoRR*, abs/2103.09092, 2021. Source code: <https://gitlab.com/ualib/ualib.gitlab.io>. arXiv:2103.09092.
- 9 William DeMeo. The Agda Universal Algebra Library, Part 3: Identity. *CoRR*, 2021. (to appear) Source code: <https://gitlab.com/ualib/ualib.gitlab.io>.
- 10 Martín Hötzel Escardó. Introduction to univalent foundations of mathematics with Agda. *CoRR*, abs/1911.00580, 2019. arXiv:1911.00580.
- 11 Emmanuel Gunther, Alejandro Gadea, and Miguel Pagano. Formalization of universal algebra in Agda. *Electronic Notes in Theoretical Computer Science*, 338:147 – 166, 2018. The 12th Workshop on Logical and Semantic Frameworks, with Applications (LSFA 2017). doi:<https://doi.org/10.1016/j.entcs.2018.10.010>.
- 12 Per Martin-Löf. An intuitionistic theory of types: predicative part. In *Logic Colloquium '73 (Bristol, 1973)*, pages 73–118. Studies in Logic and the Foundations of Mathematics, Vol. 80. North-Holland, Amsterdam, 1975.
- 13 nLab authors. Constructive Mathematics. <http://ncatlab.org/nlab/show/constructive%20mathematics>, March 2021. Revision 65.
- 14 nLab authors. Predicative Mathematics. <http://ncatlab.org/nlab/show/predicative%20mathematics>, March 2021. Revision 22.
- 15 nLab authors. Propositions as Types. <http://ncatlab.org/nlab/show/propositions%20as%20types>, March 2021. Revision 40.
- 16 Ulf Norell. Dependently typed programming in agda. In *International school on advanced functional programming*, pages 230–266. Springer, 2008.
- 17 Ulf Norell. Dependently typed programming in Agda. In *Proceedings of the 6th International Conference on Advanced Functional Programming*, AFP'08, pages 230–266, Berlin, Heidelberg, 2009. Springer-Verlag. URL: <http://dl.acm.org/citation.cfm?id=1813347.1813352>.
- 18 The Univalent Foundations Program. *Homotopy Type Theory: Univalent Foundations of Mathematics*. Lulu and The Univalent Foundations Program, Institute for Advanced Study, 2013. URL: <https://homotopytypetheory.org/book>.
- 19 Bas Spitters and Eelis Van der Weegen. Type classes for mathematics in type theory. *CoRR*, abs/1102.1323, 2011. arXiv:1102.1323.
- 20 The Agda Team. Agda Language Reference section on Axiom K, 2021. URL: <https://agda.readthedocs.io/en/v2.6.1/language/without-k.html>.
- 21 The Agda Team. Agda Language Reference section on Safe Agda, 2021. URL: <https://agda.readthedocs.io/en/v2.6.1/language/safe-agda.html#safe-agda>.
- 22 The Agda Team. Agda Tools Documentation section on Pattern matching and equality, 2021. URL: <https://agda.readthedocs.io/en/v2.6.1/tools/command-line-options.html#pattern-matching-and-equality>.
- 23 The Agda Team. The Fin type of the Agda Standard Library, 2021. URL: <https://agda.github.io/agda-stdlib/Data.Fin.html>.
- 24 Philip Wadler, Wen Kokke, and Jeremy G. Siek. *Programming Language Foundations in Agda*. July 2020. URL: <http://plfa.inf.ed.ac.uk/20.07/>.

A Dependency Graph



